
FOREWORD

MP/M II is a multi-user operating system for any microcomputer based on an 8-bit Zilog Z80 or Intel 8080 or 8085 microprocessor. Typically, an MP/M II system resides in approximately 27k. 16k of the operating system must reside in common memory.

The version of MP/M II that Digital Research ships cannot be directly booted on any specific hardware configuration. However, all the hardware dependent code is isolated in specific subroutines that can be modified by the user.

This document describes the procedures required to implement MP/M II for a custom hardware environment. At minimum, the custom hardware environment must include an 8080, 8085, or Z80 processor, 32K bytes of random access memory (RAM), a system console, and a real-time clock. This manual assumes the reader is familiar with the following Digital Research publications:

· MP/M II User's Guide

· MP/M II Programmer's Guide

it is also assumed that the reader has already implemented a CP/M 2 Basic Input Output System (BIOS), preferably on the target MP/M II machine.

1 MP/M II Alteration Procedure

The MP/M II operating system is designed so that the user can alter a specific set of subroutines that define the hardware operating environment. By modifying these subroutines, the user can produce a diskette that operates with any IBM-3740 format compatible diskette subsystem and other peripheral devices.

Although the standard MP/M II is shipped on single density floppy disks, field-alteration features allow the user to adapt MP/M II to a wide variety of disk subsystems, including single drive minidisks and high-capacity "hard disk" systems.

To achieve these independence, MP/M II has isolated all hardware-dependent code into an XIOS module. The user can rewrite the distributed version of the XIOS to customize the interface between the remaining MP/M II modules and the user's own hardware system. The user can also rewrite the distributed version of the LDRBIOS, which loads the MP/M II system from disk.

There are actually two versions of the XIOS: the RESXIOS for non-banked systems, and the BNKXIOS for banked memory systems. To avoid repeating both names for each reference, the term XIOS refers to both versions.

1.1 Preparation for MP/M II Alteration

To simplify the alteration process, this document assumes that a CP/M 2 BIOS has already been implemented on the target MP/M II machine. You must implement both the BIOS as well as the XIOS because the MP/M II loader uses a CP/M 2 BIOS to load the MP/M II system. Once loaded, MP/M II uses the XIOS and not the BIOS. The CP/M 2 BIOS used by the MP/M II loader is called the LDRBIOS.

Another good reason for implementing CP/M 2 on the target MP/M machine is that debugging your XIOS is simpler when you can run SID or DDT under a CP/M 2 system.

1.2 Customizing the MPMLDR

To customize the MPMLDR, you must integrate a LDRBIOS for your hardware configuration into the MPMLDR.COM file supplied on the distribution disk. The required LDRBIOS can be simply a version of your CP/M 2 BIOS, altered as described below and renamed to LDRBIOS.

The customized LDRBIOS must have an ORG of 1700H, perform console output functions, and be able to read data from a single disk drive. The first call MPMLDR makes to LDRBIOS is SELDSK: select disk. If your system has devices that require initialization, place initialization code or perhaps a call to the LDRBIOS cold start at the beginning of the SELDSK handler.

The LDRBIOS need only perform the operations described above. Other functions can be deleted to conserve space. There is only one restriction on memory space for LDRBIOS: it cannot extend above the base of the MPM.SYS which it is loading. (GENSYS lists MP/M II's base address in its load map.) However, if you plan to boot MP/M II from floppy disks, you will encounter a LDRBIOS upper address limit of lA00H in order to place the MPMLDR.COM file on two system tracks.

Test LDRBIOS completely to ensure that it properly performs console character output and disk reads. Be especially careful that no disk write operations occur accidently during read operations, and check that the proper track and sectors are addressed on all reads.

Use the following steps to integrate a custom LDRBIOS into the MPMLDR.COM:

1. Obtain access to a CP/M system and prepare a LDRBIOS.HEX file.

2. Read the MPMLDR.COM file into memory using either DDT or SID

A>DDT MPMLDR.COM

DDT VERS 2.0

NEXT PC

1780 0100
3. Using the input command (I), specify that the LDRBIOS.HEX file, is to be read in and then read (R) in the file. This operation overlays the LDRBIOS portion of the MP/M loader.

--ILDRBIOS.HEX

-R

NEXT PC

1A00 0000
4. Exit the debugger, returning to the CCP by executing a jump to location zero.

-GO

5. Write the updated memory image onto a disk file. Use the CP/M SAVE command to write the updated memory image onto a disk file. In the example below, the X in front of the filename simply designates an experimental version, and preserves the original.

A>SAVE 26 XMPMLDR.COM

6. Test XMPMLDR.COM and then rename it to MPMLDR.COM.

1.3 Customizing the XIOS

As you are tailoring MP/M II for your computer system, your new XIOS will require software development and testing. Two sample XIOS's are listed in the Appendixes, and can be used as models for the customized package.

The XIOS entry points, including both basic and extended, are described in Sections 2 and 3. These sections, along with the appendixes, give you the information you need to write your XIOS. Your initial implementation of an XIOS should use polled I/O without any interrupts. This initial system can run without a clock interrupt. Implement interrupts only after your XIOS is fully developed and tested.

Follow the procedure below to prepare a BNKXIOS.SPR or RESXIOS.SPR file front your customized XIOS:

1. Assemble your BNKXIOS.ASM or RESXIOS.ASM, with RMAC or any other assembler that can generate a file of type REL in Microsoft' s relocatable object file format.

A>RMAC BNKXIOS

2. Link the BNKXIOS.REL or RESXIOS.REL file using the Digital Research LINK-80 to produce the BNKXIOS.SPR or RESXIOS.SPR file.

A>LINK BNKXIOS[OS]

1.4 Debugging the XIOS

You can debug an XIOS or a resident system process with DDT or SID running under CP/M. The debugging technique is outlined in the following steps:

1. Determine the amount of memory available to MP/M II when the debugger and CP/M are resident. Do this by loading the debugger and then listing the jump instruction at location 0005H. This jump is to the base of the debugger.

A>DDT

DDT VERS 2.0

-L5

0005  JMP C6OO

2. Using GENSYS running under CP/M, generate an MPM.SYS file that specifies the top of memory determined by the previous step, allowing at least 256 bytes for a patch area.

Top page of operating system (xx) ? C6

Also while executing GENSYS, specify a breakpoint restart number different from the one used by the CP/M debugger you plan to use. The suggested MP/M II restart is #6; however, any restart from #l to #6 can usually be used. The CP/M debuggers normally use restart #7.

Breakpoint RST (xx) ? 6

Note: If you are also debugging a resident system process, be sure to select it for inclusion in MPM.SYS during GENSYS execution.

3. Using CP/M, load the MPMLDR.COM file into memory.

A>DDT MPMLDR.COM

DDT VERS 2.0

NEXT PC

1A00 0100

4. Place the characters "$B" into locations 005DH and 005EH of the default FCB based at 005CH. This operation can be done with the I command:

-I$B

The "$B" causes the MPMLDR to break after loading the MPM.SYS file. You can specify the breakpoint restart to be executed by the MPMLDR by adding one additional character to the string in the fourth position of the default FCB.

-I$B6

in the example above, a restart #6 is to be executed by the MPMLDR when loading of the MPM.SYS file is completed. If no restart number is supplied, the default restart is #7.

Remember, the restart number at the location 5FH is the CP/M debugger restart number, not the MP/M debugger restart.

5. Execute the MPMLDR.COM program by entering a G command;

-G

6. After the G command, the MP/M II loader loads the MP/M II operating system into memory and displays a, memory map. You may obtain a hard copy of your load map during the GENSYS operation by entering a, TP before executing GENSYS.

7. If you are debugging an XIOS, note the address of the BNKXIOS.SPR or RESXIOS.SPR memory segment. You must also note the address of SYSTEM.DAT. If you are debugging a resident system process, note its address as well. The debugger lists actual addresses at the console. If your hard copy listing of the XIOS or RSP starts at zero, you must add the base address listed in the GENSYS load map to each address on the listing to make the listing reflect actual addresses. Or you can assemble the code again with an additional ORG statement specifying the base listed in the load map, although the object code generated by this assembly is unusable.

8. Using the X command, determine the MP/M II beginning execution address. The address is the first location past the current program counter.

-X


.......................

P = 09F2 

In the example shown above, MP/M II execution starts at address 09F3H, which is the first instruction after the restart at 09F2H.

9. Begin execution of MP/M II using the G command, specifying the start address and any breakpoints you need in your code. The actual memory address can be determined by entering an H command to add the code segment base address given in the memory map to the relative displacement address in your XIOS or resident system process listing.

The following example shows how to set a breakpoint in an XIOS at the list subroutine entry point given in the memory map:

XIOSJMP TBL  C300H  0100H

-G9F3,C30F

09F3H is the beginning MP/M II execution address and C30FH is the XIOS jump vector address of the list subroutine.

10. At this point, you have MP/M II running with CP/M and the CP/M debugger also in memory. Because interrupts are left enabled during operation of the CP/M debugger, ensure that interrupt driven code does not execute through a breakpoint.

Because the CP/M debugger operates with interrupts left enabled, it is a somewhat difficult task to debug an interrupt driven console handler. Approach this problem by leaving console #0 in a polled mode while debugging the other consoles in an interrupt-driven mode. Once this is done, very little, if any, debugging is required to adapt the interrupt-driven code from another console to console #0. It is further recommended that you maintain a debug version of your XIOS that has polled I/O for console #0. otherwise, it is not possible to run the CP/M debugger underneath the MP/M II system because the CP/M debugger cannot get any console input, as all of it is sent to the MP/M interrupt-driven console #0 handler.

1.5 Directly Booting MP/M II

In systems where MP/M II is to be booted directly at cold start rather than loaded and run as a transient program under CP/M, the customized MPMLDR.COM file and cold start loader can be placed on the first two tracks of a eight-inch floppy disk. If a CP/M SYSGEN.COM program is available, use it to write the MPMLDR.COM file on the first two tracks. If a SYSGEN.COM program is not available, or if SYSGEN.COM does not work because a different media such as a five-inch floppy disk or hard disk is to be used, the user must write two programs: a simple memory loader, called GETSYS, which brings the MP/M loader into memory, and a program called PUTSYS, which places the MPMLDR on the first two tracks of a disk. If you have implemented a CP/M 2 BIOS, you have probably already prepared GETSYS and PUTSYS.

You can use either the SID or DDT debugger instead of writing a GETSYS program. This method is shown in the following example, which also uses SYSGEN in place of PUTSYS. Sample skeletal GETSYS and PUTSYS programs are given in Section 1.5.3.

To load and run the MP/M system automatically, you must also supply a cold start loader that loads the MP/M loader into memory from the first two tracks of the diskette. Modify the CP/M 2 cold start loader in the following manner: change the load address to 0100H and the execution address to 0100H.

The following bootstrap techniques ate specific to the Intel MDS-800, which has a boot ROM that loads the first track into location 3000H. However, the steps shown can be applied in a general sense to any custom hardware environment.

1.5.1 Preparing an MP/M II Boot Using SYSGEN

If a SYSGEN program is available, use the following steps to prepare a diskette that cold starts MP/M II:

1. Prepare the MPMLDR.COM file by integrating your custom LDRBIOS as described in Section 1.2. Test the MPMLDR.COM and verify that it operates properly.

2. Execute either DDT or SID.

A>DDT

DDT VERS 2.0

3. Using the input command (I), specify that the MPMLDR.HEX file is to be read in and then read (R) in the file with an offset of 880H bytes.

-IMPMLDR.HEX

-R880

NEXT PC

2480 0100

4. Using the I command, specify that the BOOT.HEX file is to be read in and then read in the file with an offset that loads the boot into memory at 900H. You can use the H command to calculate the offset.

-H900 3000

3900 D900

-IBOOT.HEX

-RD900

NEXT PC

2480 0000

5. Return to the CP/M console command processor (CCP) by jumping to location zero.

-G0

6. Use the SYSCEN program to write the new cold start loader onto the first two tracks of the diskette.

A>SYSGEN

SYSGEN VER 2.0

SOURCE DRIVE NAME (OR RETURN TO SKIP) <cr>

DESTINATION DRIVE NAME (OR RETURN TO REBOOT) B

DESTINATION ON B, THEN TYPE RETURN <cr>

FUNCTION COMPLETE

1.5.2 Custom Generation of an MP/M II Boot

If a SYSGEN program is not available, then use the following steps to prepare a diskette that cold starts MP/M II:

1. Write a GETSYS program that reads the custom MPMLDR.COM file into location 3380H and the cold start loader (or boot program) into location 3300H. Code GETSYS so that it starts at location 100H (base of the TPA).

Or, as in the previous example, you can use either SID or DDT to perform this function instead of writing a GETSYS program

2. Run the GETSYS program using an initialized MP/M II diskette to see if GETSYS loads the MP/M loader starting at 3380H (the operating system actually starts 128 bytes later at 3400H).

3. Write a PUTSYS program that writes memory starting at 3380H back onto the first two tracks of the diskette. PUTSYS program should be located at 200H.

4. Test the PUTSYS program using a blank, uninitialized diskette by writing a portion of memory to the first two tracks; clear memory and read it back. Test PUTSYS completely, because you will use this program to alter the MP/M II system diskette.

5. Use PUTSYS to place the MP/M II loader and cold start loader onto the first two tracks of a blank diskette.

1.5.3 Sample GETSYS and PUTSYS Programs

The following programs provide a framework for the GETSYS and PUTSYS program. You must insert WRITESEC subroutines to write the specific sectors.

;GETSYS PROGRAM - READ TRACKS 0 AND I TO MEMORY AT 3380H

;   REGISTER    USE

;       A       (SCRATCH REGISTER)

;       B       TRACK COUNT (0, 1)

;       C       SECTOR COUNT (1,2,...,26)

;       DE      (SCRATCH REGISTER PAIR)

;       HL      LOAD ADDRESS

;       SP      SET TO STACK ADDRESS

START:  LXI     SP,3380H        ;SET STACK POINTER TO SCRATCH AREA

        LXI     H, 3380H        ;SET BASE LOAD ADDRESS

        MVI     B, 0            ;START WITH TRACK 0

RDTRK:                          ;READ NEXT TRACK (INITIALLY 0)

        MVI     C,1             ;READ STARTING WITH SECTOR 1

RDSEC:                          ;READ NEXT SECTOR

        CALL    READSEC         ;USER-SUPPLIED SUBROUTINE

        LXI     D,128           ;MOVE LOAD ADDRESS TO NEXT 1/2 PAGE

        DAD     D               ;HL = HL + 128

        INR     C               ;SECTOR = SECTOR + 1

        MOV     A,C             ;CHECK FOR END OF TRACK

        CPI     27

        JC      RDSEC           ;CARRY GENERATED IF SECTOR < 27

;ARRIVE HERE AT END OF TRACK, MOVE TO NEXT TRACK

        INR     B

        MOV     A,B             ;TEST FOR LAST TRACK

        CPI     2

        JC      RDTRK           ;CARRY GENERATED IF TRACK < 2

;ARRIVE HERE AT END OF LOAD, HALT FOR NOW

        HLT

        ;USER-SUPPLIED SUBROUTINE TO READ THE DISK

READSEC:

        ;ENTER WITH TRACK NUMBER IN REGISTER B,

        ;SECTOR NUMBER IN REGISTER C, AND

        ;ADDRESS TO FILL IN HL

        PUSH    B       ;SAVE B AND C REGISTERS

        PUSH    H       ;SAVE HL REGISTERS

        ;------------------------------------------

        ;perform disk read at this point, branch to

        ;label START if an error occurs

        POP     H       ;RECOVER HL

        POP     B       ;RECOVER B AND C REGISTERS

        RET


;BACK TO MAIN PROGRAM

        END     START

;PUTSYS PROGRAM - WRITE TRACKS 0 AND I FROM MEMORY AT 3380H

;   REGISTER    USE

;       A       (SCRATCH REGISTER)

;       B       TRACK COUNT (0, 1)

;       C       SECTOR COUNT (1,2,...,26)

;       DE      (SCRATCH REGISTER PAIR)

;       HL      LOAD ADDRESS

;       SP      SET TO STACK ADDRESS

START:  LXI     SP,3380H        ;SET STACK POINTER TO SCRATCH AREA

        LXI     H, 3380H        ;SET BASE LOAD ADDRESS

        MVI     B, 0            ;START WITH TRACK 0

WRTRK:                          ;WRITE NEXT TRACK (INITIALLY 0)

        MVI     C,1             ;WRITE STARTING WITH SECTOR 1

WRSEC:                          ;WRITE NEXT SECTOR

        CALL    WRITESEC        ;USER-SUPPLIED SUBROUTINE

        LXI     D,128           ;MOVE LOAD ADDRESS TO NEXT 1/2 PAGE

        DAD     D               ;HL = HL + 128

        INR     C               ;SECTOR = SECTOR + 1

        MOV     A,C             ;CHECK FOR END OF TRACK

        CPI     27

        JC      WRSEC           ;CARRY GENERATED IF SECTOR < 27

;ARRIVE HERE AT END OF TRACK, MOVE TO NEXT TRACK

        INR     B

        MOV     A,B             ;TEST FOR LAST TRACK

        CPI     2

        JC      WRTRK           ;CARRY GENERATED IF TRACK < 2

;ARRIVE HERE AT END OF LOAD, HALT FOR NOW

        HLT

        ;USER-SUPPLIED SUBROUTINE TO WRITE THE DISK

WRITESEC:

        ;ENTER WITH TRACK NUMBER IN REGISTER B,

        ;SECTOR NUMBER IN REGISTER C, AND

        ;ADDRESS TO FILL IN HL

        PUSH    B       ;SAVE B AND C REGISTERS

        PUSH    H       ;SAVE HL REGISTERS

        ;-------------------------------------------

        ;perform disk write at this point, branch to

        ;label START if an error occurs

        POP     H       ;RECOVER HL

        POP     B       ;RECOVER B AND C REGISTERS

        RET


;BACK TO MAIN PROGRAM

        END     START

1.6 Loading MPM.SYS without the MPMLDR

The MPM.SYS file is a fully relocated absolute file that can be moved directly into memory and then executed without the use of the MPMLDR. The format of the MPM.SYS file is in Table 1-1, below.

Table 1-1. MPM.SYS File Format

Record
Contents

1
First 128 bytes of the SYSDAT page

2
Second 128 bytes of the SYSDAT page

3-n
MP/M operating system in reverse order, top down.

The actual base of the SYSDAT page in memory is specified in byte 000 of the SYSDAT page. The rest of MP/M II operating system is to be located directly below the SYSDAT page. In Table 1-1, n represents the number of records. Bytes 120-121 of the SYSDAT page contain the value of n. The execution address of MP/M is specifed by the page address given in byte 0ll of the SYSDAT page.

MPMLDR could load the MPM.SYS file into memory and then move it to its destination specified in the SYSDAT page (byte 000). Or the user could write a separate custom program to produce a directly loadable memory image from the MPM.SYS file.

1.7 Digital Research Copyright and Trademark

Read your MP/M II Licensing Agreement; it specifies your legal responsibilities when copying the MP/M II system. Place the copyright notice:

Copyright C 1981 Digital Research

on the label of each copy you make of your customized MP/M II diskette. Digital Research also requests that you place your MP/M II serial number on the label of any copies you make. Remember also that MP/M II is a trademark of Digital Research, and the first time it appears on a disk label or in a document, it should be followed by a trademark symbol, as shown below:

MP/M II, TM

1.8 Disk Organization

This section describes MP/M II sector allocation for a system in which the MPMLDR resides on the first two tracks of a single density diskette. The first sector (see Table 1-2) contains an optional software boot section. Disk controllers are often set up to bring track 0, sector 1 into memory at a specific location, often location 0000H. The program in this sector, called BOOT, is responsible for bringing the remaining sectors into memory starting at location 0100H. If your controller does not have a built-in sector load, you can ignore the program in track 0, sector 1, and begin the load from track 0 sector 2 to location 0100H.

As on example the Intel MDS-800 hardware cold start loader brings track 0, sector 1 into absolute address 3000H. When this sector is loaded, control transfers to location 3000H, where the bootstrap operation commences by loading the remainder of track 0, and all of track I into memory, starting at 0100H. Remember that this bootstrap loader is of little use in a non-MDS environment, but it is useful to examine it because you will have to duplicate some of its actions in your own cold start loader.

Table 1-2. MP/M II Sample Disk organization

Track#
Sector#
Page offset
Memory Address
MP/M Module name

00
01

(boot address)
Cold Start Loader

00
02
00
0100H
MPMLDR

:
03

0180H
:

:
04
01
0200H
:

:
05

0280H
:

:
06
02
0300H
:

:
07

0380H
:

:
08
03
0400H
:

:
09

0480H
:

:
10
04
0500H
:

:
11

0580H
:

:
12
05
0600H
:

:
13

0680H
:

:
14
06
0700H
:

:
15

0780H
:

:
16
07
0800H
:

:
17

0880H
:

:
18
08
0900H
:

:
19

0980H
:

:
20
09
0A00H
:

:
21

0A80H
:

:
22
10
0B00H
:

:
23

0B80H
:

:
24
11
0C00H
:

:
25

0C80H
MPMLDR

00
26
12
0D00H
LDRBDOS

01
01

0D80H
:

:
02
13
0E00H
:

:
03

0E80H
:

:
04
14
0F00H
:

:
05

0F80H
:

:
06
15
1000H
:

:
07

1080H
:

:
08
16
1100H
:

:
09

1180H
:

:
10
17
1200H
:

:
11

1280H
:

:
12
18
1300H
:

:
13

1380H
:

:
14
19
1400H
:

:
15

1480H
:

:
16
20
1500H
:

:
17

1580H
:

:
18
21
1600H
:

:
19

1680H
LDRBDOS

:
20
22
1700H
LDRBIOS

:
21

1780H
:

:
22
23
1800H
:

:
23

1880H
:

:
24
24
1900H
:

:
25

1980H
:

01
26
25
lA00H
LDRBIOS

2 The MP/M II BIOS

2.1 MP/M II BIOS Overview

The MP/M II BDOS and XDOS access peripheral devices as "logical" devices within the BIOS and XIOS. To customize MP/M II for a specific hardware environment, the system implementor must prepare the BIOS and XIOS subroutines upon which the BDOS and XDOS depend. This section describes how the logical portions of MP/M II expect to interact with the BIOS; Section 3 describes the same for the XIOS.

The BDOS and XDOS call BIOS subroutines through a "jump vector" located at the base of the BIOS as shown below and in Appendixes D and E. The jump vector is a sequence of 26 jump instructions that send program control to the individual BIOS subroutines. All subroutines must be represented in the jump vector during MP/M II system regeneration. However, certain subroutines may be "empty", that is, they may contain only a single RET instruction.

The BIOS jump vector must take the form shown below. The individual jump addresses for each entry point are listed to the left. Note that the XIOS entry points immediately follow the last BIOS entry point.

BIOS+00H
JMP COMMONBASE
COMMONBASE, TERMINATE PROCESS

BIOS+03H
JMP WBOOT
WARM BOOT, TERMINATE PROCESS

BIOS+06H
JMP CONST
CHECK FOR CONSOLE CHAR READY

BIOS+09H
JMP CONIN
READ CONSOLE CHARACTER IN

BIOS+0CH
JMP CONOUT
WRITE CONSOLE CHARACTER OUT

BIOS+0FH
JMP LIST
WRITE LIST CHARACTER OUT

BIOS+12H
JMP PUNCH
not used by MP/M II

BIOS+15H
JMP READER
not used by MP/M II

BIOS+18H
JMP HOME
MOVE TO TRACK 00

BIOS+1BH
JMP SELDSK
SELECT DISK DRIVE

BIOS+1EH
JMP SETTRK
SET TRACK NUMBER

BIOS+21H
JMP SETSEC
SET SECTOR NUMBER

BIOS+24H
JMP SETDMA
SET DMA ADDRESS

BIOS+27H
JMP READ
READ SELECTED SECTOR

BlOS+2AH
JMP WRITE
WRITE SELECTED SECTOR

BIOS+2DH
JMP LISTST
not used by MP/M II

BIOS+30H
JMP SECTRAN
SECTOR TRANSLATE SUBROUTINE

Each jump address corresponds to a particular subroutine that putforras a specific function, as outlined in Section 2.3. Three major functions are performed by calls to the jump table: process termination from COMMONBASE and WBOOT; simple character I/O from CONST, CONIN, CONOUT, and LIST; and disk I/O from HOME, SELDSK, SETTRK, SETSEC, SETDMA, READ, WRITE, and SECTRAN.

All simple character I/O operations are assumed to be performed in ASCII, upper and lower case, with high-order (parity) bit set to zero. The BDOS depends on only the CONST, CONIN, and CONOUT subroutines for simple character I/O. An ASCII (Z (1AH) is interpreted as an end-of-file condition for an input device.

2.2 BIOS Device Characteristics and Entry Points

The BIOS generally supports three types of devices: consoles, list devices and disks. The characteristics of each device are described below.

Consoles are the principal interactive devices that communicate with operators, and are accessed through CONST, CONIN, and CONOUT. Typically, consoles are devices such as CRTs or teletypes. MP/M II supports up to 16 console or character I/O devices.

List Devices, if they exist on your system, are usually hard copy devices, such as printers or teletypes. MP/M II supports up to 16 list devices.

Disks are accessed through a sequence of calls on the various disk I/O subroutines. These subroutines set up the disk number to access, the track and sector on a particular disk, and the direct memory access (DMA) address involved in the I/O operation. After all these parameters have been set up, a call is made to the READ or WRITE function to perform the actual I/O operation. Note that there is often a single call to SELDSK to select a disk drive, followed by a number of read or write operations to the selected disk before selecting another drive for subsequent operations. Similarly, there may be a single call to set the DMA address, followed by several calls which read or write from the selected DMA address before the DMA address is changed. The track and sector subroutines are always called before the READ or, WRITE operations are performed.

Note that the READ and WRITE routines should perform several retries (10 is standard) before reporting an error condition to the BDOS. If the error condition is returned to the BDOS, it reports the error to the user. The HOME subroutine may or may not actually perform the track 00 seek, depending upon your controller characteristics; the important point is that track 00 has been selected for the next operation, and is often treated in exactly the same manner as SETTRK with a parameter of 00.

Table 2-1 outlines the exact responsibilities of each subroutine entered through the BIOS jump table.

Table 2-1. BIOS Subroutine Summary

Subroutine
Description

COMMONBASE
The COMMONBASE entry point establishes the base address of the portion of the XIOS that must reside in common memory. The COMMONBASE entry point also contains a jump vector that enables the XIOS to access user and system memory bank switching subroutines, the MP/M II dispatcher, the XDOS and BDOS, the SYSDAT page, and COLDSTART. The effect of a call to COMMONBASE is to terminate the calling program. Other external procedures accessed by COMMONBASE are described in Section 2.4.



WBOOT
The WBOOT subroutine performs an XDOS terminate process call, terminating the calling process. The subroutine must be re-entrant and this entry point must be above the COMMONBASE label.



CONST
The CONST subroutine obtains the status of the console device specified by register D and returns 0FFH in register A if a character is ready to read, or 00H in register A if no console characters are ready. This subroutine must be re-entrant and this entry point must be above the COMMONBASE label.



CONIN
The CONIN subroutine reads the next character from the console device specified by register D into register A, and sets the parity bit (high-order bit) to zero. If no console character is ready, CONIN waits unti1 a character is typed before returning. This subroutine must be re-entrant and this entry point must be above the COMMONBASE label.



CONOUT
The CONOUT subroutine sends the character from register C to the console output device specified by register D. The character is in ASCII, with high-order parity bit set to zero. You may want to include a delay on a line feed or carriage return if your console device requires some time interval at the end of the line (such as a TI Silent 700 terminal). You can, if you wish, filter out control characters that cause your console device to react in a strange way. For example, a (Z causes the Lear-Siegler terminal to clear the screen, and could be filtered out by CONOUT. This subroutine must be re-entrant and this entry point must be above the COMMONBASE label.



LIST
The LIST subroutine sends the character from register C to the list output device specified by register D. The character is in ASCII with zero parity. This subroutine must be re-entrant and this entry point must be above the COMMONBASE label.



PUNCH
The punch device is not implemented under MP/M II. The transfer vector position is preserved to maintain CP/M compatibility. Note that MP/M II supports up to 16 character I/O devices, any of which can be a reader/punch.



READER
The reader device is not implemented under MP/M II.

See the note above for PUNCH.



HOME
The HOME subroutine returns the disk head of the currently selected disk to the track 00 position. If your controller allows access to the track 0 flag from the drive, step the head until the track 0 flag is detected.  If your controller does not support this feature, you can translate the HOME call into a call on SETTRK with a parameter of 0.



SELDSK
The SELDSK subroutine selects the disk drive given by register C for further operations, where register C contains 0 for drive A, 1 for drive B, and so up to 15 for drive P. On each disk select, SELDSK must return in HL the base address of a 16-byte area, called the Disk Parameter Header, described in Section 2.3. For standard floppy disk drives, the contents of the header and associated tables does not change, and thus the program segment included in the sample XIOS performs this operation automatically. If there is an attempt to select a non-existent drive, SELDSK returns HL=0000H as an error indicator.

On entry to SELDSK, it is possible to determine whether it is the first time the specified disk has been selected. Register E, bit 0 (least significant bit) is a zero if the drive has not been previously selected. This information is of interest in systems that read configuration information from the disk to set up a dynamic disk definition table. Although SELDSK must return the header address on each call, it is advisable to postpone the actual physical disk select operation until an I/O function (read or write) is actually performed. This is because disk selects often occur without ultimately performing any disk I/O, and many controllers unload the head of the current disk before selecting the new drive. This unloading can cause an excessive amount of noise and disk wear. The first SELDSK subroutine call that MP/M II makes is only for getting the DIRBUF address and need not perform any actual I/O.



SETTRK
For the SETTRK subroutine, register BC contains the track number for subsequent disk accesses on the currently selected drive. You can choose to seek the selected track at this time, or delay the seek until the next read or write actually occurs. Register BC can take on values in the range 0…76 corresponding to valid track numbers for standard floppy disk drives, and 0…65535 for non-standard disk subsystems.



SETSEC
For the SETSEC subroutine, register BC contains the translated sector number for subsequent disk accesses on the currently selected drive (see SECTRAN, below). You can choose to send this information to the controller at this point, or instead delay sector selection until a read or write operation occurs. Register BC can take on values in the range 1…26 corresponding to valid sector numbers for standard floppy disk drives, and 0…65535 for non-standard disk subsystems.



SETDMA
For the SETDMA subroutine, register BC contains the DMA (disk memory access) address for subsequent read or write operations. For example, if B = 00H and C = 80H when SETDMA is called, then all subsequent read operations read their data into 80H through 0FFH, and all subsequent write operations get their data from 80H through 0FFH, until the next call to SETDMA occurs. The initial DMA address is assumed to be 80H (relative to the base of the memory segment from which the call was made). Note that the controller need not actually support direct memory access. If, for example, all data is received and sent through I/O ports, the XIOS you construct can use the 128 byte area starting at the selected DMA address for the memory buffer during subsequent read or write operations.

A special case of the SETDMA subroutine occurs when the passed parameter in register BC contains a 0FFFFH. This parameter indicates that the blocking buffer, if it exists, must be flushed.

Thus, a call to the SETDMA subroutine is interpreted as a flush buffer call when a parameter of 0FFFFH is passed. The BDOS function to flush buffers is translated to this form of a SETDMA subroutine call. If the flush buffer operation performed as a result of the 0FFFFH parameter is successful a simple return should be executed. However, if a disk error occurs the current return address should be popped from the stack and one of the following error codes should be returned in the register A:

     1   -  non-recoverable error condition occurred

     2   -  disk read/only



READ
Assuming the drive has been selected, the track has been set, the sector has been set, and the DMA address has been specified, the READ subroutine attempts to read one sector based upon these parameters, and returns the following error codes in register A:

     0   -  no errors occurred

     1   -  non-recoverable error condition occurred.

If the value in register A is 0, then MP/M II assumes that the disk operation was completed properly. If an error occurs, however, the XIOS should attempt at least 10 retries to see if the error is recoverable. When an error is reported, the BDOS prints the message "BDOS ERR ON x: BAD SECTOR". Then, depending on the error mode of the calling process, the calling process is terminated or returned an error code.

An additional parameter containing
the absolute record number for the disk read is now passed by MP/M II on entry to the READ subroutine. The parameter is three bytes in length, with the high-order byte in register B and the low-order two bytes in register DE. This parameter may be useful in blocking/deblocking algorithms.

The BNKXIOS of MP/M II allows portions of the XIOS to reside in bank-switched memory (non-common). This reduces the common memory requirements. The XIOS code for all the disk operations including READ and WRITE can reside in non-common memory with one exception: the code that actually performs the transfer of data into the DMA address must reside in common memory. Two additional entry points within the XIOS, named SWTUSER and SWTSYS, enable switching between the user's memory bank and the system bank containing the BNKXIOS. SWTUSER and SWTSYS are described in Section 2.4.

If you perform deblocking in your READ and WRITE code, you must choose whether to place your deblocking buffer in common memory and then perform a single move into the user's DMA, or to place your deblocking buffer in non-common memory. If you choose the latter, you must then perform an extra move to first move the sector into common memory and then another move into the user's DMA. Blocking and deblocking are discussed in Section 2.5.



WRITE
The WRITE subroutine writes the data from the currently selected DMA address to the currently selected drive, track, and sector. The data should be marked as "non deleted data" to maintain compatibility with other CP/M and MP/M systems. WRITE returns the following error codes in register A, as shown below:

     0   -  no errors occurred

     1   -  non-recoverable error condition occurred

     2   -  disk read/only

If the value in register A is 0, then MP/M II assumes that the disk operation completed properly. If an error occurs, however, the XIOS should attempt at least 10 retries to see if the error is recover-able. When an error is reported, the BDOS prints the message "BDOS ERR ON x: BAD SECTOR". Then, depending on the error mode of the calling process, the calling process is terminated or returned an error code.

On entry to the WRITE subroutine a parameter is passed in the C register which is intended for use by blocking/deblocking algorithms. This parameter is described in Section 2.5 on blocking/deblocking.

An additional parameter containing the absolute record number for the disk write is now passed by MP/M II on entry to the WRITE subroutine. The parameter is three bytes in length, with the high-order byte in register B and the low-order two bytes in register DE. This parameter can be useful in blocking/deblocking algorithms.

See the previous section on disk READ for a discussion of placing disk WRITE code in bank-switched memory and deblocking in your WRITE code.



LISTST
The LISTST subroutine returns the ready status of the list device specified by register D. The value 00 is returned in A if the list device is not ready to accept a character, and 0FFH if a character can be sent to the printer. Note that a 00 value always suffices. LISTST must be re-entrant. This entry point is maintained solely for compatibility with CP/M and can generally be omitted from the MP/M II XIOS as none of the standard utilities use this entry point.



SECTRAN
The SECTRAN subroutine performs logical sector to physical sector translation and can improve the overall response of MP/M II. Standard MP/M II systems are shipped with a "skew factor" of 6, where six physical sectors are skipped between each logical read operation. This skew factor allows enough time between sectors for most programs to load their buffers without missing the next sector.

For computer systems that use fast processors, memory and disk subsystems, you can change the skew factor to improve overall response. Note, however, that you should maintain a single-density IBM compatible version of MP/M II for information transfer into and out of your computer system, using a skew factor of 6. In general, SECTRAN receives a logical sector number in BC and a translate table address in DE. SECTRAN uses the sector number as an index into the translate table, and returns the resulting physical sector number in HL. For standard systems, the tables and indexing code are provided in the XIOS and need not be changed.



2.3 BIOS Disk Definition Tables

This section presents the organization and construction of tables within the BIOS that define the characteristics of a particular disk system used with MP/M II. These tables can be either hand-coded or automatically generated using the DISKDEF utility provided with MP/M II. The elements of these tables are presented below.

2.3.1 Disk Parameter Table Format

In general, each disk drive has an associated (16-byte) Disk Parameter Header which both contains information about the disk drive and provides a scratchpad area for certain BDOS operations. The format of the Disk Parameter Header for each drive is shown below.

Disk Parameter Header

XLT
0000
0000
0000
DIRBUF
DPB
CSV
ALV

l6b
16b
l6b
l6b
16b
l6b
l6b
l6b

Each element is a word (16-bit) value. The meaning of each Disk Parameter Header (DPH) element is given in Table 2-2.

Table 2-2. Disk Parameter Header Elements

Element
Description

XLT
Offset of the logical to physical translation vector, if used for this particular drive, or the value 0000H if no sector translation takes place (i.e. the physical and logical sector numbers are the same). Disk drives with identical sector skew factors share the same translate tables.



0000
Scratchpad values for use within the BDOS (initial value is unimportant).



DIRBUF
Offset of a 128 byte scratchpad area for directory operations within BDOS. All DPHs address the same scratchpad area. The same DIRBUF is used by all drives.



DPB
Offset of a disk parameter block for this drive. Drives with identical disk characteristics address the same disk parameter block.



CSV
Offset of a scratchpad area used for software check for changed disks. This offset is different for each DPH.



ALV
Offset of a scratchpad area used by the BDOS to keep disk storage allocation information. This offset is different for each DPH.



Given n disk drives, the DPHs are arranged in a table whose first row of 16 bytes corresponds to drive 0, with the last row corresponding to drive n-l. The table thus appears as:

DPBASE

00    XLT00   0000  0000  0000  DIRBUF  DPB00  CSV00  ALV00

01    XLT01   0000  0000  0000  DIRBUF  DPB01  CSV01  ALV01

n-l   XLTn-1  0000  0000  0000  DIRBUF  DPBn-1 CSVn-1 ALVn-l

where the label DPBASE defines the offset of the DPH table relative to the beginning of the operating system.

A responsibility of the SELDSK subroutine, defined in the previous section, is to return the offset of the DPH from the beginning of the operating system for the selected drive. The following sequence of operations returns the table offset, with a 0000H returned if the selected drive does not exist.

NDISKS  EQU     4       ;NUMBER OF DISK DRIVES

SELDSK:         ;SELECT DISK N GIVEN BY C

        LXI     H,0000H ;READY FOR ERR

        MOV     A,C

        CPI     NDISKS  ;N BEYOND MAX DISKS?

        RNC             ;RETURN IF SO

                        ;0 <= N < NDISKS

        MOV     L,C

        DAD     H       ;READY FOR * 16

        DAD     H

        DAD     H

        DAD     H

        LXI     D,DPBASE

        DAD     D       ;DPBASE + N * 16

        RET

The translation vectors (XLT 00 through XLTn-1) are located elsewhere in the BIOS, and simply correspond one-for-one with the logical sector numbers zero through the sector count-1. The Disk Parameter Block (DPB) for each drive is more complex. A particular DPB, which is addressed by one or more DPHs, takes the general form:

SPT
BSH
BLM
EXM
DSM
DRM
AL0
AL1
CKS
OFF

l6b
8b
8b
8b
l6b
l6b
8b
8b
16b
l6b

where each is a byte or word value, as shown by the "8b" or "16b" indicator below the field. The fields are defined in Table 2-3.

Table 2-3. Disk Parameter Block Fields

Field
Definition

SPT
is the total number of sectors per track.



BSH
is the data allocation block shift factor, determined by the data block allocation size.



BLM
is the block mask which is also determined by the data block allocation size.



EXM
is the extent mask, determined by the data block allocation size and the number of disk blocks.



DSM
determines the total storage capacity of the disk drive.



DRM
determines the total number of directory entries which can be stored on this drive.



AL0, AL1
determine reserved directory blocks.



CKS
is the size of the directory check vector, a CKS of 8000H marks the drive as permanent with no directory records checked.



OFF
is the number of reserved tracks at the beginning of the (logical) disk.



Although these table values are produced automatically by DISKDEF, it is worthwhile reviewing the derivation of each field so that the values may be cross-checked when necessary. The values of BSH and BLM determine (implicitly) the data allocation size BLS, which is not an entry in the disk parameter block. Given that you have selected a value for BLS, the values of BSH and BLM are shown in Table 2-4 below, where all values are in decimal.

Table 2-4. BSH and BLM values for Selected BLS

BLS
BSH
BLM

1,024
3
7

2,048
4
15

4,096
5
31

8,192
6
63

16,384
7
127

The value of EXM depends upon both the BLS and whether the DSM value is less than 256 or greater than 255, as shown in the following table.

Table 2-5. Maximum EXM Values

BLS
DSM < 256
DSM > 255

1,024
0
N/A

2,048
1
0

4,096
3
1

8,192
7
3

16,384
15
7

The value of DSM is the maximum data block number supported by this particular drive, measured in BLS units. The product BLS times (DSM+l) is the total number of bytes held by the drive and, of course, must be within the capacity of the physical disk, not counting the reserved operating system tracks.

The DRM entry is one less than the total number of directory entries, which can take on a 16-bit value. The values of AL0 and AL1, however, are determined by DRM. The two values AL0 and AL1 can together be considered a string of 16-bits, as shown below.

AL0
AL1

00   01   02   03   04   05   06   07
08   09   10   11   12   13   14   15

where position 00 corresponds to the high-order bit of the byte labeled AL0, and 15 corresponds to the low-order bit of the byte labeled AL1. Each bit position reserves a data block for a number of directory entries, thus allowing a total of 16 data blocks to be assigned for directory entries (bits are assigned starting at 00 and filled to the right until position 15). Each directory entry occupies 32 bytes, as shown in Table 2-6.

Table 2-6. BLS and Number of Directory Entries

BLS
Directory Entries

1,024
32 times # bits

2,048
64 times # bits

4,096
128 times # bits

8,192
256 times # bits

16,384
512 times # bits

Thus, if DRM = 127 (128 directory entries), and BLS = 1024, then there are 32 directory entries per block, requiring 4 reserved blocks. In this case, the 4 high-order bits of AL0 are set, resulting in the values AL0 = 0F0H and AL1 = 00H.

The CKS value is determined as follows: if the disk drive media is removable, then CKS = (DRM+1)/4, where DRM is the last directory entry number. If the media is fixed, then set CKS = 8000H (no directory records are checked in this case and drive marked as permanent).

Finally, the OFF field determines the number of tracks which are skipped at the beginning of the physical disk. This value is automatically added whenever SETTRK is called, and can be used as a mechanism for skipping reserved operating system tracks, or for partitioning a large disk into smaller segmented sections.

To complete the discussion of the DPB, recall that several DPHs can address the same DPB if their drive characteristics are identical. Further, the DPB can be dynamically changed when a new drive is addressed by simply changing the pointer in the DPH since the BDOS copies the DPB values to a local area whenever the SELDSK function is invoked.

Returning back to the DPH for a particular drive, note that the two address values CSV and ALV remain. Both addresses reference an area of uninitialized memory following the BIOS. The areas must be unique for each drive, and the size of each area is determined by the values in the DPB.

The size of the area addressed by CSV is CKS bytes, which is sufficient to hold the directory check information for this particular drive. If CKS = (DRM+1)/4, then you must reserve (DRM+1)/4 bytes for directory check use. If CKS = 0, indicating no checked directory entries, or CKS = 8000H, marking the drive as permanent with no checked directory entries, then no storage is reserved.

The size of the area addressed by ALV is determined by the maximum number of data blocks allowed for this particular disk, and is computed as (DSM/8)+l.

2.3.2 The DISKDEF Macro Library

A macro library called DISKDEF greatly simplifies the table construction process. You must have access to the MAC macro assembler or the RMAC relocatable macro assembler distributed with MP/M II to use the DISKDEF facility. The macro library is included with all MP/M II distribution disks.

A BIOS disk definition consists of the following sequence of macro statements:

MACLIB  DISKDEF

DISKS   n

DISKDEF 0,...

DISKDEF 1,...

DISKDEF n-1

ENDEF

where the MACLIB statement loads the DISKDEF.LIB file (on the same disk as your BIOS) into MACs internal tables. The DISKS macro call follows, which specifies the number of drives to be configured with your system, where n is an integer in the range I to 16. A series of DISKDEF macro calls then follow, which define the characteristics of each logical disk, 0 through n-1 (corresponding to logical drives A through P). Note that the DISKS and DISKDEF macros generate the in-line fixed data tables described in the previous section, and thus must be placed in a non-executable portion of your BIOS, typically directly following the BIOS jump vector.

The remaining portion of your BIOS is defined following the DISKDEF macros, with the ENDEF macro call immediately preceding the END statement. The ENDEF (End of Diskdef) macro generates the necessary uninitialized RAM areas that are located in memory above your BIOS.

The form of the DISKDEF macro call is

DISKDEF dn,fsc,lsc,[skf],bls,dks,dir,cks,ofs,[kl6],[prm]

where

dn
is the 'Logical disk number, 0 to n-1

fsc
is the first physical sector number (0 or 1)

lsc
is the last sector number

skf
is the optional sector skew factor

bls
is the data allocation block size

dks
is the total number of blocks on the drive.

dir
is the number of directory entries

cks
is the number of "checked" directory entries

ofs
is the track offset to logical track 00

k16
is an optional 1.4 compatibility flag which forces 16K/directory entry

prm
is an optional flag which indicates that the drive is permanent (cannot be removed)

The value dn is the drive number being defined with this DISKDEF macro invocation. The fsc parameter accounts for differing sector numbering systems, and is usually 0 or 1. The lsc is the last numbered sector on a track. When present, the skf parameter defines the sector skew factor which is used to create a sector translation table according to the skew. If the number of sectors is less than 256, a single-byte table is created, otherwise each translation table element occupies two bytes. No translation table is created if the skf parameter is omitted (or equal to 0).

The b1s parameter specifies the number of bytes allocated to each data block, and takes on the values 1024, 2048, 4096, 8192, or 16384. Generally, performance increases with larger data block sizes since there are fewer directory references and logically connected data records are physically close on the disk. Also each directory entry addresses more data, and the BIOS-resident RAM space is reduced. The dks specifies the total disk size in b1s units. That is, if the bls = 2048 and dks = 1000, then the total disk capacity is 2,048,000 bytes. If dks is greater than 255, then the block size parameter bls must be greater than 1024. The value of dir is the total number of directory entries which may exceed 255, if desired.

The cks parameter determines the number of directory items to check on each directory scan and is used internally to detect changed disks during system operation. When this situation is detected, MP/M II automatically marks the disk read/only, so that data is not subsequently destroyed. As stated in the previous section, the value of cks equals dir when the media is easily changed, as is the case with a floppy disk subsystem. If the disk is permanently mounted, then the value of cks is typically 0 and thus the prm parameter should be included to indicate that the drive is permanent.

The ofs value determines the number of tracks to skip when this particular drive is addressed, which can be used to reserve additional operating system space or to simulate several logical drives on a single large-capacity physical drive.

The k16 parameter is included when file compatibility is required with versions of CP/M 1.4 that have been modified for higher density disks. This parameter ensures that only 16K is allocated for each directory record, as was the case for previous versions. Normally, this parameter is left null. Finally, the prm parameter can be used to indicate that the drive is permanent. This parameter should only be included if the disk media cannot be removed from the drive.

For convenience and economy of table space, the special form

DISKDEF i,j

gives disk i the same characteristics as a previously defined drive j. A standard four-drive single density system, which is compatible with CP/M 1.4, is defined using the following macro invocations:

DISKS   4

DISKDEF 0,1,26,6,1024,243,64,64,2

DISKDEF 1,0

DISKDEF 2,0

DISKDEF 3,0

ENDEF

with all disks having the same parameter values of 26 sectors per track (numbered 1 through 26), with 6 sectors skipped between each access, 1024 bytes per data block, 243 data blocks for a total of 243k byte disk capacity, 64 checked directory entries, and two operating system tracks.

The DISKS macro generates n Disk Parameter Headers (DPHs) starting at the DPH table address DPBASE generated by the macro. Each disk header block contains sixteen bytes, as described above, and corresponds one-for-one to each of the defined drives. In the four drive standard system, for example, the DISKS macro generates a table of the form:

DPBASE  EQU     $

DPE0:   DW      XLT0,0000H,0000H,0000H,DIRBUF,DPB0,CSV0,ALV0

DPE1:   DW      XLT0,0000H,0000H,0000H,DIRBUF,DPB0,CSV1,ALV1

DPE2:   DW      XLT0,0000H,0000H,0000H,DIRBUF,DPB0,CSV2,ALV2

DPE3:   DW      XLT0,0000H,0000H,0000H,DIRBUF,DPB0,CSV3,ALV3

where the DPH labels are included for reference purposes to show the beginning table addresses for each drive, 0 through 3. The values contained within the disk parameter header are described in detail in the previous section. The check and allocation vector addresses are generated by the ENDEF macro in the RAM area following the BIOS code and tables.

Note that if the skf (skew factor) parameter is omitted (or equal to 0), the translation table is omitted, and a 0000H value is inserted in the XLT position of the disk parameter header for the disk. In a subsequent call to perform the logical to physical translation, SECTRAN receives a translation table address of DE = 0000H, and simply returns the original logical sector from BC in the HL register pair. A translate table is constructed when the skf parameter is present, and the (non-zero) table address is placed into the corresponding DPHs. The table shown below, for example, is constructed when the standard skew factor skf = 6 is specified in the DISKDEF macro call:

XLT0:   DB      1,7,13,19,25,5,11,17,23,3,9,15,21

        DB      2,8,14,20,26,6,12,18,24,4,10,16,22

Following the ENDEF macro call, a number of uninitialized data areas are defined. These data areas need not be a part of the BIOS that is loaded upon cold start, but must be available between the BIOS and the end of memory. The size of the uninitialized RAM area is determined by EQU statements generated by the ENDEF macro. For a standard four-drive system, the ENDEF macro might produce:

4C72 =  BEGDAT EQU $

          (data areas)

4DB0 =  ENDDAT EQU $

013C =  DATSIZ EQU $-BEGDAT

which indicates that uninitialized RAM begins at location 4C72H, ends at 4DB0H-1, and occupies 013CH bytes. You must ensure that these addresses are free for use after the system is loaded.

After modification, you can use the STAT program to check your drive characteristics, because STAT uses the disk parameter block to decode the drive information. The STAT command form

STAT d:DSK:

decodes the disk parameter block for drive d (d=A, … P) and displays the values shown below.

r:
128 Byte Record Capacity

k:
Kilobyte
 Drive Capacity

d:
32 Byte Directory Entries

c:
Checked Directory Entries

e:
Records/
Extent

b:
Records/
Block

s:
Sectors/Track

t:
Reserved Tracks

Three examples of DISKDEF macro invocations are shown below with corresponding STAT parameter values. The last example produces an 8-megabyte system.

DISKDEF 0,1,58,,2048,256,128,128,2

r=4096, k=512, d=128, c=128, e=256, b=16, s=58, t=2

DISKDEF 0,1,58,,2048,1024,300,0,2

r=16384, k=2048, d=300, c=0, e=128, b=16, s=58, t=2

DISKDEF 0,1,58,,16384,512,128,128,2

r=65536, k=8192, d=128, c=128, e=1024, b=128, s=58, t=2

2.4 External Procedure Access

To help the XIOS access other MP/M entry points, a jump vector is dynamically built by the MP/M II GENSYS program and placed at the COMMONBASE subroutine entry point. The dynamic portion of the jump vector contains five entry points that provide access to user and system memory bank switching, the MP/M II dispatcher, the XDOS, and the SYSDAT page. Table 2-7 describes external procedure entry points.

The following example illustrates the code used to access external procedures:

COMMONBASE: JMP     COLDSTART

SWTUSER:    JMP     $-$

SWTSYS:     JMP     $-$

PDISP:      JMP     $-$

XDOS:       JMP     $-$

SYSDAT:     DW      $-$

COLDSTART:

WBOOT:

            mvi     c,0

            jmp     XDOS    ;terminate process

Table 2-7. External Procedure Summary

Subroutine
Description

SWTUSER
The SWTUSER entry point restores the bank of the user's calling program. There are no parameters passed or returned. The purpose of SWTUSER is to enable BIOS disk read and write code to transfer data from a disk controller or buffer in common memory to/from the DMA buffer in the user's calling program. This procedure must be called only from common memory, that is above the COMMONBASE label, and it must be used only from BIOS disk functions. Internally the SWTUSER procedure disables and then re-enables interrupts. Thus, if you disable interrupts before calling SWTUSER they will be enabled on returning from SWTUSER.



SWTSYS
The SWTSYS entry point restores the bank of the BNKBDOS. There are no parameters passed or returned. The purpose of SWTSYS is to restore the bank containing the banked portion of the BDOS following the transfer of data from a disk controller or buffer in common memory to/from the DMA buffer in the user's calling program. This procedure must be called only from common memory. Internally the SWTSYS procedure disables and then re-enables interrupts. Thus, if you disable interrupts before calling SWTSYS they will be enabled on returning from SWTSYS.



PDISP
The PDISP entry point forces a dispatch call. It is intended to be used at the conclusion of interrupt handling when a process is to be dispatched. It is effectively a null procedure call from the point of view of the calling program.



XDOS
The XDOS entry point provides access to XDOS functions. XDOS functions are required for flag operations, queue operations and polling devices.



SYSDAT
The SYSDAT entry is not a true entry point, but the address of the system data page. Section 4 provides a definition of the system data page.



2.5 Blocking and Deblocking Algorithms

Upon each call to the BIOS WRITE entry point, the BDOS includes information that allows effective sector blocking and deblocking where the host disk subsystem has a sector size which is a multiple of the basic 128-byte unit. This section presents a general-purpose algorithm that can be included within your BIOS that uses the BDOS information to perform the operations automatically.

Upon each call to WRITE, the BDOS provides the following information in register C:

0   -  deferred write sector

1   -  non-deferred write sector

2   -  deferred write to the first sector of a new data block

3   -  non-deferred write to the first sector of a new data block

Conditions 0 and 2 occur only for permanent drives and allow deferred writes. Conditions 1 and 3 occur for non-permanent (removable) drives and force immediate (non-deferred) writes. Condition 1 also occurs on permanent drives for writes to the directory.

Conditions 2 and 3 occur when a write operation is made to the first sector of a new data block. The blocking/deblocking algorithm does not perform physical record pre-reads if sequential writes are made to a new data block. In most cases, application programs read or write multiple 128-byte sectors in sequence, and thus there is little overhead involved in either operation when blocking and deblocking records because pre-read operations can be avoided when writing records.

The blocking and deblocking algorithm is listed in Appendix B in skeletal form. The file is included on your MP/M II disk.

Generally, the algorithms map all MP/M II sector read operations onto the host disk through an intermediate buffer which is the size of tile host disk sector. Throughout the program, values and variables which relate to the sector involved in a seek operation are prefixed by "sek," while those related to the host disk system are prefixed by "hst." The equate statements beginning on line 24 define the mapping between MP/M II and the host system, and must be changed if other than the sample host system is involved.

The SELDSK entry point clears the host buffer flag whenever a new disk is logged-in. Note that although the SELDSK entry point computes and returns the Disk Parameter Header address, it does not physically select the host disk at this point (it is selected later at READHST or WRITEHST). Further, SETTRK, SETSEC, and SETDMA simply store the values, but do not take any other action at this point. SECTRAN performs a trivial function of returning the physical sector number.

The principal entry points are READ and WRITE. These subroutines take the place of your previous READ and WRITE operations.

The actual physical read or write takes place at either WRITEHST or READHST, where all values have been prepared: hstdsk is the host disk number, hsttrk is the host track number, and hstsec is the host sector number (which may require translation to a physical sector number). You must insert code at this point which performs the full host sector read or write into, or out of, the buffer at hstbuf of length hstsiz. All other mapping functions are performed by the algorithms.

2.6 Common Memory Portion of the BNKXIOS

Take care when selecting which XIOS code is to be placed in common memory. This section should give you some helpful guidelines.

In general, all XIOS and BIOS entries (with the exception of the disk I/O entries) must be above the COMMONBASE subroutine entry point. Thus, the BNKXIOS enables you to place your disk drivers in a portion of code that is not in common memory. There are, however, some exceptions that affect both the code and data areas of the disk handlers.

The Disk Parameter Headers and Disk Parameter Blocks must be in common memory.

The DIRBUF data structure, which is referenced by the disk parameter blocks, must reside in common memory.

All disk device polling code and interrupt handlers must reside in common memory.

While it is possible to place a deblocking buffer in non-common memory, it requires a sector buffer in common memory and an extra move of 128 bytes to move the data first into common memory and then into the users DMA buffer. Also, bank switching cannot be permitted while a physical DMA from a disk controller to a deblocking buffer in non-common memory is in operation.

3 The MP/M II XIOS

3.1 MP/M II XIOS Overview

The Extended Input/Output System (XIOS) must include the hardware dependent code that polls devices, handles interrupts and performs memory management functions.

The MP/M II system implementor must prepare subroutines that perform the functions described in Table 3-1, then place a jump vector containing the XIOS entry points immediately following the BIOS jump vector. Most of the XIOS subroutines need to be re-entrant. The XIOS jump vector must take the following form:

BIOS+33H
JMP SELMEMORY
SELECT MEMORY

BIOS+36H
JMP POLLDEVICE
POLL DEVICE

BIOS+39H
JMP STARTCLOCK
START CLOCK

BIOS+3CH
JMP STOPCLOCK
STOP CLOCK

BIOS+3FH
JMP EXITREGION
EXIT CRITICAL REGION

BIOS+42H
JMP MAXCONSOLE
MAXIMUM CONSOLE NUMBER

BIOS+45H
JMP SYSTEMINIT
SYSTEM INITIALIZATION

BIOS+48H
JMP IDLE
IDLE PROCEDURE (Optional)

3.2 MP/M XIOS Entry Points

Each jump address corresponds to a particular subroutine that performs the specific function. Table 3-1 outlines the exact responsibilities of each XIOS entry point subroutine.

Table 3-1. XIOS Subroutine Summary

Subroutine
Function

SELMEMORY
The SELMEMORY subroutine identifies the segment of memory where a process is to execute. Each time a process is dispatched for execution, the operating system makes a call to this XIOS select memory procedure. If the hardware environment has memory bank select ion/protect ion, SELMEMORY can use the passed parameter to select/protect areas of memory. The passed parameter (in registers BC) is a pointer to a memory descriptor from which the memory base, size, attributes and bank of the executing process can be determined. Thus, all other regions of memory can be write-protected.

MP/M II calls SELMEMORY with interrupts disabled from within the dispatcher. The SELMEMORY subroutine must not enable interrupts. This subroutine must reside above the COMMONBASE entry point.



POLLDEVICE
A polled environment can be created by coding XIOS device poll handlers. The purpose of implementing a polled environment is to avoid typical busy-wait code for device operation completion. There are also peripheral devices that may not operate efficiently under interrupts. XDOS calls the device poll handler (POLLDEVICE) with the device to be polled in the C register as a single parameter.

The user-written POLLDEVICE procedure can be coded to access the device polling routines via a table that contains the addresses of the device polling procedures.

An association is made between a device number to be polled and the polling procedure itself. The polling procedures must return a value of 0FFH in the accumulator if the device is ready, or 00H if the device is not ready. POLLDEVICE is called from a critical region within the dispatcher; therefore, the POLLDEVICE subroutine must not enable interrupts. This subroutine must reside above the COMMONBASE entry point.



STARTCLOCK
The STARTCLOCK and STOPCLOCK procedures eliminate unnecessary overhead for the system clock interrupt handler. The system clock provides a time base for both the real time flag and the system tick procedure. However, the system tick procedure is needed only when there is a process on the delay list. MP/M II calls STARTCLOCK when a process enters the delay list to initiate the system tick time base (see Section 3.4).

In some hardware environments, it is not possible to shut off the system time unit clock while maintaining the one-second flag used for keeping time of day. In this situation, the STARTCLOCK procedure simply sets a boolean variable to true, indicating that there is a delayed process. The clock interrupt handler can then determine if system time unit flag is to be set by testing the boolean. This subroutine must reside above the COMMONBASE entry point.



STOPCLOCK
When the system delay list is emptied, MP/M II calls the STOPCLOCK procedure to stop the system tick time base. This eliminates unnecessary overhead for the system clock interrupt handler.

In some hardware environments, it
is not possible to shut off the system time unit clock while maintaining the one second flag used for keeping time of day; that is, a single clock/timer interrupt source is used. In this situation, the STOPCLOCK procedure simply sets a boolean variable to false, indicating that there are no delayed processes. The clock interrupt handler can then determine if the system time unit flag is to be set by testing the boolean. This subroutine must reside above the COMMONBASE entry point.



EXITREGION
MP/M II calls the EXITREGION procedure to test a local parameter called the PREEMPT flag. If PREEMPT is true, EXITREGION leaves interrupts disabled. If PREEMPT is false, EXITREGION enables interrupts. Interrupt service routines must set the PREEMPT flag true at beginning of the interrupt handling. This procedure allows an interrupt service routine to make a flag set MP/M II system call, leaving interrupts disabled until completion of the interrupt handling. This subroutine must reside above the COMMONBASE entry point.



MAXCONSOLE
The maximum console procedure enables the calling program to determine the number of physical consoles the BIOS is capable of supporting. The number of physical consoles is returned in the A register. This subroutine must reside above the COMMONBASE entry point.



SYSTEMINIT
The system initialization procedure performs the required MP/M cold start initialization. The following is a typical initialization for a banked system: first, MP/M II initializes bank 0, disables interrupts and calls SYSTEMINIT. Then, SYSTEMINIT sets up interrupt jump vectors, interrupt masks, and the base page of each bank before returning to MP/M II. Finally, MP/M II enables interrupts. A typical initialization for a non-banked system would perform the same steps, but only one bank would be initialized. MP/M II disables interrupts and calls the SYSTEMINIT entry point prior to any other XIOS call. As stated above, MP/M II enables interrupts immediately upon return from SYSTEMINIT. This subroutine must reside above the COMMONBASE entry point. In systems with bank switched memory, it is necessary to set up the base page (0000H - 00FFH) within each bank of memory. Both the MPMLDR and MP/M itself assume that the base bank (bank #0) is switched in when the MPMLDR is executed. The base bank is properly initialized by MP/M prior to entering SYSTEMINIT. The information required for the initialization of other banks is provided on entry to SYSTEMINIT in the registers defined below:

C
MP/M debugger restart #

DE
MP/M entry point address for the debugger. Place a jump at the proper debugger restart location to the address contained in DE.

HL
BIOS direct jump table address. Place a jump instruction at location 0000H in each bank's base page to the address contained in HL.



IDLE
An IDLE process is the anchor of the process ready list. The MP/M II nucleus calls the IDLE procedure when there are no other processes ready to run. The normal IDLE procedure is a call to the dispatcher. This most efficiently serves polled devices. If your system is entirely interrupt-driven (i.e. no polled devices), you can supply your own IDLE procedure, which should be as follows:

    IDLE:   HALT

            RET

If you do not supply an IDLE procedure, place three bytes of zero at the BIOS+48H location.



3.3 Interrupt Service Routines

The MP/M II operating system is designed to work with virtually any interrupt architecture, be it flat or vectored. The code operating at the interrupt level saves the required registers, determines the cause of the interrupt, removes the interrupting condition, sets an appropriate flag, and then forces a dispatch to take place.

Be sure to use a minimum number of stack levels when saving the state of the interrupted process. This is because the interrupted application program, especially if it has been written for a CP/M environment, is not likely to provide extra stack area as a contigency for interrupts. The example Extended Input/Output Systems shown in the Appendixes illustrate a technique whereby no additional levels of stack are required beyond that of the interrupt restart itself. This technique is highly recommended.

Operation of the flags is described in Section 3 of the MP/M II Programmer's Guide, under the discussion of the Flag Set and Flag Wait XDOS Functions. Briefly, flags synchronize a process to an asynchronous event. In general, an interrupt service routine sets a particular flag while another process waits for the flag to be set.

At a logical level above the physical interrupts, the flags can be regarded as providing 256 levels of virtual interrupts (32 flags are supported under MP/M II). Thus, logical interrupt handlers wait on flags set by the physical interrupt handlers. This mechanism allows a common XDOS to operate on potentially all 8080, 8085 and Z80® microcomputers, regardless of the hardware environment.

As an example, consider a hardware environment with a flat interrupt structure. That is, a single interrupt level is provided and devices must be polled to determine the cause of the interrupt. Once the interrupt cause is determined, a specific flag is set indicating that that particular interrupt has occurred.

At the conclusion of the interrupt processing, a jump should be made to the MP/M II dispatcher. This is done by jumping to the PDISP entry point. This jump gives the processor resource to the highest priority ready process, usually the process readied by setting the flag in the interrupt handler, and then enables interrupts before jumping to resume execution of that process.

The only XDOS or BDOS call that should be made from an interrupt handler is 133: Flag Set. Any other XDOS or BDOS call results in a dispatch which would then enable interrupts before the execution of the interrupt handler is completed.

It is recommended that interrupts be used only for asynchronous operations such as console input or disk operation complete. In general, operations such as console output should not be interrupt driven, because the system has more elasticity when performing polled console outputs while idling, rather than incurring the dispatch overhead for each character transmitted. This is particularly true at higher baud rates.

If a system requires the execution of a return from interrupt (RETI) instruction, the interrupt handler must execute the RETI before branching to the dispatcher via the PDISP entry point.

3.4 Time Base Management

The XIOS must provide two time bases: a one second flag for real time and a system tick for managing the delay list. The one second flag operation is logically separate from the system tick operation even though it may physically share the same clock/timer interrupt source. The one second flag procedure sets flag #2 at each one second of real time. MP/M II uses flag #2 to maintain a time of day clock.

The system tick procedure, when enabled by STARTCLOCK, sets flag #1 at system time unit intervals. The recommended time unit is a period of 16.67 milliseconds, corresponding to a tick frequency of 60 Hz. When operating with 50 Hz, use a 20 millisecond period. MP/M II uses the system tick to manage the delay list until the delay list is empty, at which time the system tick procedure is disabled by STOPCLOCK.

The system tick frequency is critical because it determines the dispatch frequency for compute-bound processes. If the frequency is too high, a significant amount of system overhead is incurred by excessive dispatches. If the frequency is too low, compute-bound processes keep the CPU resource for accordingly longer periods.

4 MP/M II System File Components

The MP/M II system file, MPM.SYS, consists of a number of components: the system data page, the customized XIOS, the RESBDOS and BNKBDOS, the XDOS and BNKXDOS, the TMP, and the resident system processes. MPM.SYS resides in the directory with a user code of 0 and usually has the Read Only attribute. The MP/M II loader reads the MPM.SYS file into memory to bring up the MP/M II system.

4.1 System Data

The system data page contains 256 bytes used by GENSYS to dynamically configure the MP/M II system. The system data page can be prepared using the GENSYS program or it can be manually prepared using DDT or SID. The Table 4-1 describes the byte assignments.

Table 4-1. System Data Byte Assignments

Byte
Contents

000-000
Mem$top, top page of memory

001-001
Nmb$cns, number of system consoles (TMPs)

002-002
Brkpt$RST, breakpoint RST #

003-003
Add system call user stacks, boolean

004-004
Bank switched, boolean

005-005
Z80 version, boolean

006-006
banked BDOS, boolean

007-007
XIOS jump table page

008-008
RESBDOS base page

009-010
CP/NET master configuration table address

011-011
XDOS base page

012-012
RSPs (BNKXIOS top+l) base page

013-013
BNKXIOS base page

014-014
BNKBDOS base page

015-015
Max$mem$seg, max memory segment number

016-047
Initial memory segment table

048-063
Breakpoint vector table, filled in by debuggers

064-079
Reserved for MP/M II

080-095
System call user stack pointer table

096-119
Reserved for MP/M II

120-121
Number of records in MPM.SYS file

122-122
# ticks/sec

123-123
System Drive

124-124
Common Memory Base Page

125-125
Number of RSPs

126-127
Listcp array Address

128-143
Subflg, submit flag array

144-186
Reserved for MP/M II

187-187
Max locked records/process

188-188
Max open files/process

189-190
# list items

191-192
Pointer to base of lock table free space

193-193
Total system locked records

194-194
Total system open files

195-195
Dayfile logging, boolean

196-196
Temporary file drive

197-197
Number of printers

197-241
Reserved for MP/M II

242-242
Banked XDOS base page

243-243
TMP process descriptor base

244-244
Console.dat base

245-246
BDOS/XDOS entry point

247-247
TMP.SPR base

248-248
Nmbrsps, number of banked RSPs

249-249
Brsp base address

250-251
Brspl, non-resident rsp process link

252-253
Sysdatadr, XDOS internal data segment address

254-255
Rspl, resident system process link

4.2 Customized XIOS

The customized XIOS is obtained either from a file named RESXIOS.SPR, or a file named BNKXIOS.SPR. The XIOS file of type SPR contains the page relocatable version of the user-customized XIOS. The standard method for the generation of the XIOS is to use the Digital Research LINK program. An alternative method is described in Section 1.

4.3 BDOS

The Basic Disk Operating System (BDOS) resides in two page relocatable files named the RESBDOS and the BNKBDOS. These two files contain the console, list and disk file management code.

4.3.1 RESBDOS

The file named RESBDOS.SPR is a page relocatable file containing the logical console and list handling, as well as the resident portion of the disk file system that provides an interface to the BNKBDOS.

4.3.2 BNKBDOS

The file named BNKBDOS.SPR is a page relocatable file containing the non-resident portion of the banked BDOS.

4.4 XDOS

The XDOS file named XDOS.SPR is a page relocatable file containing the priority-driven MP/M II nucleus. The nucleus contains the following code pieces: root module, dispatcher, queue management, flag management, memory management, terminal handler, terminal message process, command line interpreter, file name parser, and time base management.

4.5 Resident System Processes

A file type of RSP identifies a resident system process. The RSP files distributed with MP/M II include: run-time system status display (MPMSTAT), printer spooler (SPOOL), abort named process (ABORT), and a scheduler (SCHED). At system generation time, GENSYS prompts you to select which RSPs to include in the MPM.SYS file.

It is possible for the user to prepare custom resident system processes. The resident system processes must follow these rules:

· The file must be page-relocatable. Page relocatable files can be generated by LINK, or by the submit files MACSPR.SUB or ASMSPR.SUB. The output file must be renamed to type RSP.

· The first two bytes of the resident system process are reserved for the address of the BDOS/XDOS. Thus a resident system process can access the BDOS/XDOS by loading the two bytes at relative 0000-0001H and then performing a PCHL.

· The process descriptor for the resident system process must begin at the third byte position.

4.6 Banked Resident System Processes

A banked resident system process consists of two parts: a resident portion and the code for the process. The resident portion contains the process descriptor, and queues or other data structures that must be in common memory. This portion follows the rules given above for resident system processes. The presence of a banked portion is specified by setting the process descriptor memory segment index to zero rather than 0FFH. The name provided in the process descriptor is used to obtain the banked portion which has a file type of BRS.

The second part of a banked system process is the actual code piece for the process. The rules for the BRS portion are as follows:

· The file must be page relocatable. Page relocatable files can be generated by LINK, or the procedure outlined in Section 1. The output file must be renamed to type BRS.

· Bytes 0000-0001H of the banked RSP are reserved for the address of the resident portion of the RSP. Thus, a banked RSP must access the BDOS/XDOS functions by indirectly loading from the two bytes at relative 0000-0001H, which point to the base of the resident portion of the RSP, which in turn contain the BDOS/XDOS entry point address.

· Bytes 0002-0003H of the banked RSP must contain the initial stack pointer value for the process. Thus the stack for the banked RSP is in the banked portion of the RSP, and should be initialized such that the return address on top of the stack is the banked RSP entry point address.

· Bytes 0004-000BH of the banked RSP must contain an ASCII name for the process. This is used for display purposes during GENSYS and MPMLDR execution.

5 System Generation

5.1 GENSYS Operation

MP/M II system generation consists of preparing a system data file and concatenating both required and optional code files to produce a file named MPM.SYS. A GENSYS program reforms these tasks and can be run under either MP/M II or CP/M. The GENSYS automates the system generation process by prompting the user for optional parameters and then prepares the MPM.SYS file. The following sample execution illustrates GENSYS operation.

0A>gensys

MP/M-80 V2.0 System Generation

Copyright (C) 1981, Digital Research

Default entries are shown in (parens).

Default base is Hex, precede entry with # for decimal

Use SYSTEM.DAT for defaults (Y) ?

Top page of operating system (FF) ?

Number of TMPs (system consoles) (#2) ?

Number of Printers (#1) ?

Breakpoint RST (06) ?

Add system call user stacks (Y) ?

Z80 CPU (Y) ?

Number of ticks/second (#60) ?

System Disk (E:) ?

Temporary file drive (E:) ?

Maximum locked records/process (#16) ?

Total locked records/system (#32) ?

Maximum open files/process (#l6) ?

Total open files/system (#32) ?

Bank switched memory (Y) ?

Number of user memory segments (#3) ?

Common memory base page (C0) ?

Dayfile logging at console (Y) ?

SYSTEM  DAT FF00H 0100H

TMPD    DAT FE00H 0100H

USERSYS STK FD00H 0100H

XIOSJMP TBL FC00H 0100H

Accept new system data page entries (Y) ?

RESBDOS SPR F000H 0C00H

XDOS    SPR CE00H 2200H

Select Resident System Processes:

SCHED   RSP (N) ?

ABORT   RSP (N) ? y

SPOOL   RSP (N) ? y

MPMSTAT RSP (N) ? y

ABORT   RSP CD00H 0100H

SPOOL   RSP CC00H 0100H

MPMSTAT RSP CB00H 0100H

BNKXIOS SPR B800H 1300H

BNKBDOS SPR 9500H 2300H

BNKXDOS SPR 9200H 0300H

TMP     SPR 8F00H 0300H

SPOOL   BRS 8700H 0800H

MPMSTAT BRS 7900H 0E00H

LCKLSTS DAT 7700H 0200H

CONSOLE DAT 7500H 0200H

Enter memory segment table:

Base,size,attrib,bank (75,8B,80,00) ?

Base,size,attrib,bank (00,C0,00,01) ?

Base,size,attrib,bank (00,C0,00,02) ?

Base,size,attrib,bank (00,C0,00,03) ? 00,ff,0,0

*** Memory conflict - segment trimmed

Base,size,attrib,bank (00,75,00,00) ?

MP/M II Sys 7500H 8B00H Bank 00

Memseg  Usr 0000H C000H Bank 01

Memseg  Usr 0000H C000H Bank 02

Memseg  Usr 0000H 7500H Bank 00

Accept new memory segment table entries (Y) ?

** GENSYS DONE **

5.2 System Generation Parameters

This section discusses the issues involved in answering each of the GENSYS queries shown in the example above.

5.2.1 Defaults

The GENSYS program displays default entry values within parentheses. The base is hex unless a # character preceeds the value to indicate a decimal base. The initial prompt determines if the internal GENSYS defaults are to be used, or those of the most recently generated SYSTEM.DAT file.

5.2.2 Top Page of Operating System

Enter two hex ASCII digits to give the top page of the operating system. The highest address used by MP/M II is XXFFH, where XX is the entry.

5.2.3 Number of System Consoles

This entry determines the number of system consoles for which Terminal Message Processes (TMP's) are created to generate user prompts and send command lines to the Command Line Interpreter (CLI). A region of common memory called TMPD.DAT is reserved for the TMP process descriptors. Four TMP process descriptors can be placed in each page of the TMPD.DAT. Each system console also requires 256 bytes of memory for stack and buffer areas in a non resident region of memory called CONSOLE.DAT. MP/M II supports up to a maximum of 16 character I/O console devices, of which 8 can be system consoles and have associated TMPs. During MP/M II initialization, an XIOS call obtains the actual maximum number of physical consoles supported by the XIOS. This number is used if it is less than the number specified during the GENSYS.

5.2.4 Number of Printers

This entry determines the number of physical printers which the XIOS is capable of supporting. This number is used by the MPMSTAT program when it displays the status of the system printers.

5.2.5 Breakpoint RST

Enter the breakpoint restart number to be used by the MP/M debuggers. Recommended restarts are RST #1 to RST #6.

5.2.6 System Call User Stacks

If you want to execute CP/M *.COM files, enter yes. An affirmative response forces a stack switch to occur when system calls are made from a user program. BDOS calls require more stack space under MP/M II than under CP/M. An affirmative response causes GENSYS to allocate a region of common memory called USERSYS.STK. The size of this region is determined by the number of user memory segments, where 0-3 segments require 100h bytes and 4-7 segments require 200h bytes.

Note that this affects BDOS calls only, not XDOS calls. The XDOS is re-entrant and performs no stack switching. Therefore, if your program makes any XDOS calls, you need to make certain that you have allocated sufficient stack.

5.2.7 Z80 CPU

An affirmative response should only be made if you do have a Z80 CPU. If specified, the MP/M II dispatcher saves and restores the Z80 alternate register set.

5.2.8 Number of Ticks / Second

This entry value can be used by applications programs to determine the number of ticks per second. This value may vary among MP/M II systems.

5.2.9 System Disk

The drive entered here is used for a second search if the file requested to the CLI is not found on the default drive.

5.2.10 Temporary File Drive

The drive entered here is used as the drive for temporary disk files. This entry is used by SUBMIT when it generates the $$$.SUB temporary file. This entry can also be accessed in the system data page by application programs as the drive on which to create temporary files.

5.2.11 Maximum Locked Records / Process

This entry specifies the maximum number of records that a single process (usually one program) can lock at any given time. This number can range from 0 to 255 and must be less than or equal to the total locked records for the system.

5.2.12 Total Locked Records / System

This entry specifies the total number of locked records for all the processes executing under MP/M II at any given time. This number can range from 0 to 255 and should be greater than or equal to the maximum locked records per process.

It is possible to allow each process to either use up the total system lock record space, or to allow each process to lock only a fraction of the system total. The first technique implies a dynamic storage region in which one process can force other processes to block because it has consumed all available resources.

5.2.13 Maximum Open Files / Process

This entry specifies the maximum number of files that a single process (usually one program) can open at any given time. This number can range from 0 to 255 and must be less than or equal to the total open files for the system.

5.2.14 Total Open Files / System

This entry specifies the total number of open files for all the processes executing under MP/M II at any given time. This number can range from 0 to 255 and should be greater than or equal to the maximum open files per process.

It is possible either to allow each process to use up the total system open file space, or to allow each process to only open a fraction of the system total. The first technique implies a dynamic storage region in which one process can force other processes to block because it has consumed all available resources.

5.2.15 Bank Switched Memory

If your system does not have bank-switched memory, then you should respond with a "No". otherwise respond with a "Y" and additional questions and responses (as shown in Section 5.2.2) are required.

5.2.16 Number of User Memory Segments

The number of user memory segments must be in the range 1 to 7 and should be greater than or equal the number of system consoles.

5.2.17 Common Memory Base Page

In response to this prompt, enter the address of the lowest page of memory common to all banks. GENSYS checks that all modules requiring residence in common memory are located above this address.

5.2.18 Dayfile Logging at Console

An affirmative response causes the generated MP/M II system to display the current time, file name and type, and user number of each executed command file.

5.2.19 Accept System Data Page Entries

If the entries made for the first 16 queries are acceptable, then enter yes. Otherwise, any or all of the entries made can be changed by re-cycling through the GENSYS queries, entering a carriage return where values are not to be changed.

5.2.20 Select Resident System Processes

GENSYS searches the directory for all files of type RSP. Each file found is listed and included in the generated system file if you respond with a "Y". Tests are performed to make certain that the specified RSPF reside at or above the common base address.

5.2.21 Memory Segment Table

Memory segmentation is defined by the entries which are made. You are prompted for the base, size, attributes, and bank for each memory segment. The GENSYS program only allows you to enter the number of segments specified in the response to the query regarding the number of user memory segments.

The first default entry made is for the operating system. This becomes the segment zero entry in the memory segment table. It is switched in during the banked MP/M II execution of the BNKXIOS, BRS's, and the BNKBDOS. The first entry is not counted in your number of user memory segments.

A significant amount of error checking is performed using a memory bit map to ensure that no memory segments overlap each other. It will be possible to customize the GENSYS program such that non existent memory for a particular hardware configuration is pre-allocated in the bit map.

The order of entries in the memory segment table is also critical. The first entry is reserved for the operating system. The remaining entries can be specified by user. In specifying the user memory segments, the absolute TPA regions (segments based at 0000H) should be specified in order of size, from the largest to the smallest. Entering the segments in this order causes the MP/M II memory manager to allocate the largest available TPA region for execution by a COM program because it linearly searches through the memory segment table for the first available segment based at zero. The ordering of relocatable segments (those not based at 0000H) is not critical because the MP/M II memory manager does a best fit for those segments.

The attribute byte is normally defined as 00. However, if you wish to pre-allocate a memory segment, specify a value of FFH.

The bank byte value is an index which can be used by the XIOS to obtain a value to be sent to the bank switching hardware to select the specified bank. Values of 0,1,2,... are used to identify the memory banks. A bank byte value of 0 is used for the non resident portion of MP/M II.

5.2.22 Accept Memory Segment Table

A negative response to this query allows memory segment entries to be re-edited prior to acceptance.

5.3 GENSYS Execution

The GENSYS program has an automatic mode which simplifies repetitive generation of MPM.SYS files. This is useful in a debug mode of testing, XIOS editing, and a subsequent GENSYS execution to produce a new MPM.SYS file. The automatic mode is specified as follows:

0A>GENSYS $A

The effect of the automatic mode is to simulate the entry of a <cr> for each GENSYS query.

6 MP/M LOADER

6.1 MP/M Loader Operation and Display

The MPMLDR program loads the MPM.SYS file and branches to the execution address of the MP/M II operating system. MPMLDR can be run under CP/M or loaded from the first two tracks of a disk by the cold start loader.

The MPMLDR displays system loading and configuration. It does not require any operator interaction. In the following example, the MPM.SYS file prepared by the first GENSYS example shown in Section 5 is loaded into memory and executed.

MP/M-II V2.0 Loader

Copyright (C) 1981, Digital Research

Nmb of consoles  = 2

Breakpoint RST # = 6

Z80 Alternate register set saved/restored by dispatcher

Memory Segment Table:

SYSTEM  DAT FF00H 0100H

TMPD    DAT FE00H 0100H

USERSYS STK FD00H 0100H

XIOSJMP TBL FC00H 0100H

RESBDOS SPR F000H 0C00H

XDOS    SPR CE00H 2200H

ABORT   RSP CD00H 0100H

SPOOL   RSP CC00H 0100H

MPMSTAT RSP CB00H 0100H

BNKXIOS SPR B800H 1300H

BNKBDOS SPR 9500H 2300H

BNKXDOS SPR 9200H 0300H

TMP     SPR 8F00H 0300H

Spool   BRS 8700H 0800H

Mpmstat BRS 7900H 0E00H

LCKLSTS DAT 7700H 0200H

CONSOLE DAT 7500H 0200H

-------------------------

MP/M II Sys 7500H 8B00H Bank 0

Memseg  Usr 0000H C000H Bank 1

Memseg  Usr 0000H C000H Bank 2

Memseg  Usr 0000H 7500H Bank 0

MP/M II V2.0

Copyright (C) 1981, Digital Research

0A>

6.2 MPMLDR Execution

Two parameters may be specified to the MPMLDR. The first parameter is used to cause a break to a CP/M debugger after the loading is completed. The parameter is a $Bn character string placed in the default FCB filename field beginning at 005DH. The character n is the CP/M debugger restart number. If n is not entered, a default of 7 is used. An example of this parameter is shown in Section 1.4.

The second parameter can specify an alternate filename for loading other than the standard MPM.SYS file. This parameter is specified by placing a filename with a filetype of SYS in the default FCB beginning at 005CH, or, if the $Bn parameter is also being specified, in the second default FCB beginning at 006CH. A good application of this second parameter would be to incorporate a menu-driven SYS file selection in the LDRBIOS at the SELDSK entry point. Thus, the operator would be prompted to select the appropriate SYS file for his MP/M environment. Custom code at the SELDSK entry point would prompt the operator for a file name and then place the selected SYS file name into the default FCB beginning at 005CH.

Appendix A: Disk Definition Macro

;       MP/M II V2.0 disk re-definition library

;

;       Copyright (c) 1979, 1980, 1981

;       Digital Research

;       Box 579

;       Pacific Grove, CA

;       93950

;

;       MP/M II logical disk drives are defined using the

;       macros given below, where the sequence of calls

;       is:

;

;       disks  n

;       diskdefparameter-list-0

;       diskdefparameter-list-1

;       ...

;       diskdefparameter-list-n

;       endef

;

;       where n is the number of logical disk drives attached

;       to the MP/M II system, and parameter-list-i defines the

;       characteristics of the ith drive (i=0,1,...,n-1)

;

;       each parameter-list-i takes the form

;               dn,fsc,lsc,[skf],bls,dks,dir,cks,ofs,[k16],[prm]

;       where

;       dn      is the disk number 0,1,...,n-1

;       fsc     is the first sector number (usually 0 or 1)

;       lsc     is the last sector number on a track

;       skf     is optional "skew factor" for sector translate

;       bls     is the data block size (1024,2048,...,16384)

;       dks     is the disk size in bls increments (word)

;       dir     is the number of directory elements (word)

;       cks     is the number of dir elements to checksum

;       ofs     is the number of tracks to skip (word)

;       k16     is an optional 0 which forces 16K/directory entry

;       prm     is an optional 0 which marks drive as permanent

;

;       for convenience, the form

;               dn,dm

;       defines disk dn as having the same characteristics as

;       a previously defined disk dm.

;

;       a standard four drive MP/M II system is defined by

;               disks   4

;               diskdef 0,1,26,6,1024,243,64,64,2

;       dsk     set     0

;               rept    3

;       dsk     set     dsk+1

;               diskdef %dsk,0

;               endm

;               endef

;

;       the value of "begdat" at the end of assembly defines the

;       beginning of the uninitialize ram area above the bios,

;       while the value of "enddat" defines the next location

;       following the end of the data area.  the size of this

;       area is given by the value of "datsiz" at the end of the

;       assembly.  note that the allocation vector will be quite

;       large if a large disk size is defined with a small block

;       size.

;

dskhdr  macro   dn

;;      define a single disk header list

dpe&dn: dw      xlt&dn,0000h    ;translate table

        dw      0000h,0000h     ;scratch area

        dw      dirbuf,dpb&dn   ;dir buff,parm block

        dw      csv&dn,alv&dn   ;check, alloc vectors

        endm

;

disks   macro   nd

;;      define nd disks

ndisks  set     nd      ;;for later reference

dpbase  equ     $       ;base of disk parameter blocks

;;      generate the nd elements

dsknxt  set     0

        rept    nd

        dskhdr  %dsknxt

dsknxt  set     dsknxt+1

        endm

        endm

;

dpbhdr  macro   dn

dpb&dn  equ     $               ;disk parm block

        endm

;

ddb     macro   data,comment

;;      define a db statement

        db      data            comment

endm

;

ddw     macro   data,comment

;;      define a dw statement

        dw      data            comment

endm

;

gcd     macro  m,n

;;      greatest common divisor of m,n

;;      produces value gcdn as result

;;      (used in sector translate table generation)

gcdm    set     m       ;;variable for m

gcdn    set     n       ;;variable for n

gcdr    set     0       ;;variable for r

        rept    65535

gcdx    set     gcdm/gcdn

gcdr    set     gcdm - gcdx*gcdn

        if      gcdr = 0

        exitm

        endif

gcdm    set     gcdn

gcdn    set     gcdr

        endm

        endm

;

diskdef macro   dn,fsc,lsc,skf,bls,dks,dir,cks,ofs,k16,prm

;;      generate the set statements for later tables

        if      nul lsc

;;      current disk dn same as previous fsc

dpb&dn  equ     dpb&fsc;equivalent parameters

als&dn  equ     als&fsc;same allocation vector size

css&dn  equ     css&fsc;same checksum vector size

xlt&dn  equ     xlt&fsc;same translate table

        else

cksz    set     (cks)/4

secmax  set     lsc-(fsc)      ;;sectors 0...secmax

sectors set     secmax+1;;number of sectors

als&dn  set     (dks)/8;;size of allocation vector

        if      ((dks) mod 8) ne 0

als&dn  set     als&dn+1

        endif

css&dn  set     cksz    ;;number of checksum elements

;;      generate the block shift value

blkval  set     bls/128 ;;number of sectors/block

blkshf  set     0       ;;counts right 0's in blkval

blkmsk  set     0       ;;fills with 1's from right

        rept    16      ;;once for each bit position

        if      blkval=1

        exitm

        endif

;;      otherwise, high order 1 not found yet

blkshf  set     blkshf+1

blkmsk  set     (blkmsk shl 1) or 1

blkval  set     blkval/2

        endm

;;      generate the extent mask byte

blkval  set     bls/1024       ;;number of kilobytes/block

extmsk  set     0      ;;fill from right with 1's

        rept    16

        if      blkval=1

        exitm

        endif

;;      otherwise more to shift

extmsk  set     (extmsk shl 1) or 1

blkval  set     blkval/2

        endm

;;      may be double byte allocation

        if      (dks) > 256

extmsk  set     (extmsk shr 1)

        endif

;;      may be optional [0] in last position

        if      not nul k16

extmsk  set     k16

        endif

;;      now generate directory reservation bit vector

dirrem  set     dir    ;;# remaining to process

dirbks  set     bls/32 ;;number of entries per block

dirblk  set     0      ;;fill with 1's on each loop

        rept    16

        if      dirrem=0

        exitm

        endif

;;      not complete, iterate once again

;;      shift right and add 1 high order bit

dirblk  set     (dirblk shr 1) or 8000h

        if      dirrem > dirbks

dirrem  set     dirrem-dirbks

        else

dirrem  set     0

        endif

        endm

        dpbhdr  dn      ;;generate equ $

        ddw     %sectors,<;sec per track>

        ddb     %blkshf,<;block shift>

        ddb     %blkmsk,<;block mask>

        ddb     %extmsk,<;extnt mask>

        ddw     %(dks)-1,<;disk size-1>

        ddw     %(dir)-1,<;directory max>

        ddb     %dirblk shr 8,<;alloc0>

        ddb     %dirblk and 0ffh,<;alloc1>

        if      nul prm

        ddw     %(cks)/4,<;check size>

        else

        ddw     8000h+cksz,<;permanent disk with check size>

        endif

        ddw     %ofs,<;offset>

;;      generate the translate table, if requested

        if      nul skf

xlt&dn  equ     0               ;no xlate table

else

        if      skf = 0

xlt&dn  equ     0               ;no xlate table

        else

;;      generate the translate table

nxtsec  set     0       ;;next sector to fill

nxtbas  set     0       ;;moves by one on overflow

        gcd     %sectors,skf

;;      gcdn = gcd(sectors,skew)

neltst  set     sectors/gcdn

;;      neltst is number of elements to generate

;;      before we overlap previous elements

nelts   set     neltst  ;;counter

xlt&dn  equ     $               ;translate table

        rept    sectors ;;once for each sector

        if      sectors < 256

        ddb     %nxtsec+(fsc)

        else

        ddw     %nxtsec+(fsc)

        endif

nxtsec  set     nxtsec+(skf)

        if      nxtsec >= sectors

nxtsec  set     nxtsec-sectors

        endif

nelts   set     nelts-1

        if      nelts = 0

nxtbas  set     nxtbas+1

nxtsec  set     nxtbas

nelts   set     neltst

        endif

        endm

        endif   ;;end of nul fac test

        endif   ;;end of nul bls test

        endm

;

defds   macro   lab,space

lab:    ds      space

        endm

;

lds     macro   lb,dn,val

        defds   lb&dn,%val&dn

        endm

;

endef   macro

;;      generate the necessary ram data areas

begdat  equ     $

dirbuf: ds      128     ;directory access buffer

dsknxt  set     0

        rept    ndisks  ;;once for each disk

        lds     alv,%dsknxt,als

        lds     csv,%dsknxt,css

dsknxt  set     dsknxt+1

        endm

enddat  equ     $

datsiz  equ     $-begdat

force:  db      0       ;force out last byte in hex file

        endm

;

Appendix B: Sector Deblocking Algorithms

;*********************************************************

;*                                                       *

;*      Sector Deblocking Algorithms for MP/M II V2.0    *

;*                                                       *

;*********************************************************

;

;       utility macro to compute sector mask

smask   macro   hblk

;;      compute log2(hblk), return @x as result

;;      (2 ** @x = hblk on return)

@y      set     hblk

@x      set     0

;;      count right shifts of @y until = 1

        rept    8

        if      @y = 1

        exitm

        endif

;;      @y is not 1, shift right one position

@y      set     @y shr 1

@x      set     @x + 1

        endm

        endm

;

;*****************************************************

;*                                                   *

;*         MP/M to host disk constants               *

;*                                                   *

;*****************************************************

blksiz  equ     2048            ;MP/M allocation size

hstsiz  equ     512             ;host disk sector size

hstspt  equ     20              ;host disk sectors/trk

hstblk  equ     hstsiz/128      ;MP/M sects/host buff

cpmspt  equ     hstblk * hstspt ;MP/M sectors/track

secmsk  equ     hstblk-1        ;sector mask

        smask   hstblk          ;compute sector mask

secshf  equ     @x              ;log2(hstblk)

;

;*****************************************************

;*                                                   *

;*        BDOS constants on entry to write           *

;*                                                   *

;*****************************************************

wrall   equ     0               ;write to allocated

wrdir   equ     1               ;write to directory

wrual   equ     2               ;write to unallocated

;

;*****************************************************

;*                                                   *

;*      The BDOS entry points given below show the   *

;*      code which is relevant to deblocking only.   *

;*                                                   *

;*****************************************************

;

;       DISKDEF macro, or hand coded tables go here

dpbase  equ     $               ;disk param block base

;

boot:

wboot:

        ;enter here on system boot to initialize

        xra     a               ;0 to accumulator

        sta     hstact          ;host buffer inactive

        sta     unacnt          ;clear unalloc count

        ret

;

home:

        ;home the selected disk

        lda     hstwrt          ;check for pending write

        ora     a

        jnz     homed

        sta     hstact          ;clear host active flag

homed:

        ret

;

seldsk:

        ;select disk

        mov     a,c             ;selected disk number

        sta     sekdsk          ;seek disk number

        mov     l,a             ;disk number to HL

        mvi     h,0

        rept    4               ;multiply by 16

        dad     h

        endm

        lxi     d,dpbase        ;base of parm block

        dad     d               ;hl=.dpb(curdsk)

        ret

;

settrk:

        ;set track given by registers BC

        mov     h,b

        mov     l,c

        shld    sektrk          ;track to seek

        ret

;

setsec:

        ;set sector given by register c

        mov     a,c

        sta     seksec          ;sector to seek

        ret

;

setdma:

        ;set dma address given by BC

        mov     h,b

        mov     l,c

        shld    dmaadr

        ret

;

sectran:

        ;translate sector number BC

        mov     h,b

        mov     l,c

        ret

;

;*****************************************************

;*                                                   *

;*      The READ entry point takes the place of      *

;*      the previous BIOS defintion for READ.        *

;*                                                   *

;*****************************************************

read:

        ;read the selected MP/M sector

        xra     a

        sta     unacnt          ;unacnt = 0

        inr     a

        sta     readop          ;read operation

        sta     rsflag          ;must read data

        mvi     a,wrual

        sta     wrtype          ;treat as unalloc

        jmp     rwoper          ;to perform the read

;

;*****************************************************

;*                                                   *

;*      The WRITE entry point takes the place of     *

;*      the previous BIOS defintion for WRITE.       *

;*                                                   *

;*****************************************************

write:

        ;write the selected MP/M sector

        xra     a               ;0 to accumulator

        sta     readop          ;not a read operation

        mov     a,c             ;write type in c

        sta     wrtype

        ani     wrual           ;write unallocated?

        jz      chkuna          ;check for unalloc

;

;       write to unallocated, set parameters

        mvi     a,blksiz/128    ;next unalloc recs

        sta     unacnt

        lda     sekdsk          ;disk to seek

        sta     unadsk          ;unadsk = sekdsk

        lhld    sektrk

        shld    unatrk          ;unatrk = sectrk

        lda     seksec

        sta     unasec          ;unasec = seksec

;

chkuna:

        ;check for write to unallocated sector

        lda     unacnt          ;any unalloc remain?

        ora     a

        jz      alloc           ;skip if not

;

;       more unallocated records remain

        dcr     a               ;unacnt = unacnt-1

        sta     unacnt

        lda     sekdsk          ;same disk?

        lxi     h,unadsk

        cmp     m               ;sekdsk = unadsk?

        jnz     alloc           ;skip if not

;

;       disks are the same

        lxi     h,unatrk

        call    sektrkcmp       ;sektrk = unatrk?

        jnz     alloc           ;skip if not

;

;       tracks are the same

        lda     seksec          ;same sector?

        lxi     h,unasec

        cmp     m               ;seksec = unasec?

        jnz     alloc           ;skip if not

;

;       match, move to next sector for future ref

        inr     m               ;unasec = unasec+1

        mov     a,m             ;end of track?

        cpi     cpmspt          ;count MP/M sectors

        jc      noovf           ;skip if no overflow

;

;       overflow to next track

        mvi     m,0             ;unasec = 0

        lhld    unatrk

        inx     h

        shld    unatrk          ;unatrk = unatrk+1

;

noovf:

        ;match found, mark as unnecessary read

        xra     a               ;0 to accumulator

        sta     rsflag          ;rsflag = 0

        jmp     rwoper          ;to perform the write

;

alloc:

        ;not an unallocated record, requires pre-read

        xra     a               ;0 to accum

        sta     unacnt          ;unacnt = 0

        inr     a               ;1 to accum

        sta     rsflag          ;rsflag = 1

;

;*****************************************************

;*                                                   *

;*      Common code for READ and WRITE follows       *

;*                                                   *

;*****************************************************

rwoper:

        ;enter here to perform the read/write

        xra     a               ;zero to accum

        sta     erflag          ;no errors (yet)

        lda     seksec          ;compute host sector

        rept    secshf

        ora     a               ;carry = 0

        rar                     ;shift right

        endm

        sta     sekhst          ;host sector to seek

;

;       active host sector?

        lxi     h,hstact        ;host active flag

        mov     a,m

        mvi     m,1             ;always becomes 1

        ora     a               ;was it already?

        jz      filhst          ;fill host if not

;

;       host buffer active, same as seek buffer?

        lda     sekdsk

        lxi     h,hstdsk        ;same disk?

        cmp     m               ;sekdsk = hstdsk?

        jnz     nomatch

;

;       same disk, same track?

        lxi     h,hsttrk

        call    sektrkcmp       ;sektrk = hsttrk?

        jnz     nomatch

;

;       same disk, same track, same buffer?

        lda     sekhst

        lxi     h,hstsec        ;sekhst = hstsec?

        cmp     m

        jz      match           ;skip if match

;

nomatch:

        ;proper disk, but not correct sector

        lda     hstwrt          ;host written?

        ora     a

        cnz     writehst        ;clear host buff

;

filhst:

        ;may have to fill the host buffer

        lda     sekdsk

        sta     hstdsk

        lhld    sektrk

        shld    hsttrk

        lda     sekhst

        sta     hstsec

        lda     rsflag          ;need to read?

        ora     a

        cnz     readhst         ;yes, if 1

        xra     a               ;0 to accum

        sta     hstwrt          ;no pending write

;

match:

        ;copy data to or from buffer

        lda     seksec          ;mask buffer number

        ani     secmsk          ;least signif bits

        mov     l,a             ;ready to shift

        mvi     h,0             ;double count

        rept    7               ;shift left 7

        dad     h

        endm

;       hl has relative host buffer address

        lxi     d,hstbuf

        dad     d               ;hl = host address

        xchg                    ;now in DE

        lhld    dmaadr          ;get/put MP/M data

        mvi     c,128           ;length of move

        lda     readop          ;which way?

        ora     a

        jnz     rwmove          ;skip if read

;

;       write operation, mark and switch direction

        mvi     a,1

        sta     hstwrt          ;hstwrt = 1

        xchg                    ;source/dest swap

;

rwmove:

        ;C initially 128, DE is source, HL is dest

        ldax    d               ;source character

        inx     d

        mov     m,a             ;to dest

        inx     h

        dcr     c               ;loop 128 times

        jnz     rwmove

;

;       data has been moved to/from host buffer

        lda     wrtype          ;write type

        ani     wrdir           ;to directory?

        lda     erflag          ;in case of errors

        rz                      ;no further processing

;

;       clear host buffer for directory write

        ora     a               ;errors?

        rnz                     ;skip if so

        xra     a               ;0 to accum

        sta     hstwrt          ;buffer written

        call    writehst

        lda     erflag

        ret

;

;*****************************************************

;*                                                   *

;*      Utility subroutine for 16-bit compare        *

;*                                                   *

;*****************************************************

sektrkcmp:

        ;HL = .unatrk or .hsttrk, compare with sektrk

        xchg

        lxi     h,sektrk

        ldax    d               ;low byte compare

        cmp     m               ;same?

        rnz                     ;return if not

;       low bytes equal, test high 1s

        inx     d

        inx     h

        ldax    d

        cmp     m       ;sets flags

        ret

;

;*****************************************************

;*                                                   *

;*      WRITEHST performs the physical write to      *

;*      the host disk, READHST reads the physical    *

;*      disk.                                        *

;*                                                   *

;*****************************************************

writehst:

        ;hstdsk = host disk #, hsttrk = host track #,

        ;hstsec = host sect #. write "hstsiz" bytes

        ;from hstbuf and return error flag in erflag.

        ;return erflag non-zero if error

        ret

;

readhst:

        ;hstdsk = host disk #, hsttrk = host track #,

        ;hstsec = host sect #. read "hstsiz" bytes

        ;into hstbuf and return error flag in erflag.

        ret

;

;*****************************************************

;*                                                   *

;*      Unitialized RAM data areas                   *

;*                                                   *

;*****************************************************

;

sekdsk: ds      1               ;seek disk number

sektrk: ds      2               ;seek track number

seksec: ds      1               ;seek sector number

;

hstdsk: ds      1               ;host disk number

hsttrk: ds      2               ;host track number

hstsec: ds      1               ;host sector number

;

sekhst: ds      1               ;seek shr secshf

hstact: ds      1               ;host active flag

hstwrt: ds      1               ;host written flag

;

unacnt: ds      1               ;unalloc rec cnt

unadsk: ds      1               ;last unalloc disk

unatrk: ds      2               ;last unalloc track

unasec: ds      1               ;last unalloc sector

;

erflag: ds      1               ;error reporting

rsflag: ds      1               ;read sector flag

readop: ds      1               ;1 if read operation

wrtype: ds      1               ;write operation type

dmaadr: ds      2               ;last dma address

hstbuf: ds      hstsiz          ;host buffer

;

;*****************************************************

;*                                                   *

;*      The ENDEF macro invocation goes here         *

;*                                                   *

;*****************************************************

        end

Appendix C: Sample MP/M II Loader BIOS

title  'MP/M II V2.0 Skeleton Ldrbios'

;       Copyright (C) 1978, 1979, 1980, 1981

;       Digital Research

;       Box 579, Pacific Grove

;       California, 93950

false   equ     0

true    equ     not false

        org     1700h

buff    equ     0080h   ;default buffer address

;       jump vector for individual routines

        jmp     boot

wboote: jmp     wboot

        jmp     const

        jmp     conin

        jmp     conout

        jmp     list

        jmp     punch

        jmp     reader

        jmp     home

        jmp     seldsk

        jmp     settrk

        jmp     setsec

        jmp     setdma

        jmp     read

        jmp     write

        jmp     list$st         ; list status poll

        jmp     sect$tran       ; sector translation

boot:

wboot:

gocpm:

        ret

crtin:                  ; crt: input

        ret

crtout:                 ; crt: output

        ret

crtst:                  ; crt: status

        ret

ttyin:                  ; tty: input

        ret

ttyout:                 ; tty: output

        ret

lptout:                 ; lpt: output

        ret

lpt$st:

        ret

conin   equ     crtin

const   equ     crtst

conout  equ     crtout

reader  equ     ttyin

punch   equ     ttyout

list    equ     lptout

listst  equ     lptst

seldsk: ;select disk given by register c

        ret

;

home:   ;move to home position

        ret

;

settrk: ;set track number given by c

        ret

;

setsec: ;set sector number given by c

        ret

;

setdma: ;set dma address given by regs b,c

        ret

;

sect$tran:        ; translate the sector # in <c> if needed

        ret

;

read:   ;read next disk record (assuming disk/trk/sec/dma set)

        ret

;

write:  ;disk write function

        ret

;

        end

Appendix D: Simple XIOS Source Listing

title   'MP/M II V2.0  DSC-2 Basic & Extended I/O Systems'

        cseg

        maclib  diskdef

;

; bios for micro-2 computer

;

;

false   equ     0

true    equ     not false

;

debug   equ     true

ldcmd   equ     true

;

MHz4    equ     true

        if      MHz4

dlycnst equ     086h

        else

dlycnst equ     054h

        endif

;

;       org     0000h

;

;       jump vector for individual subroutines

;       jmp     coldstart       ;cold start

        jmp     commonbase

wboot:

        jmp     warmstart       ;warm start

        jmp     const           ;console status

        jmp     conin           ;console character in

        jmp     conout          ;console character out

        jmp     list            ;list character out

        jmp     rtnempty        ;punch not implemented

        jmp     rtnempty        ;reader not implemented

        jmp     home            ;move head to home

        jmp     seldsk          ;select disk

        jmp     settrk          ;set track number

        jmp     setsec          ;set sector number

        jmp     setdma          ;set dma address

        jmp     read            ;read disk

        jmp     write           ;write disk

        jmp     pollpt          ;list status

        jmp     sectran         ;sector translate

        jmp     selmemory       ; select memory

        jmp     polldevice      ; poll device

        jmp     startclock      ; start clock

        jmp     stopclock       ; stop clock

        jmp     exitregion      ; exit region

        jmp     maxconsole      ; maximum console number

        jmp     systeminit      ; system initialization

        db      0               ; force use of internal dispatch @ idle

;       jmp     idle            ; idle procedure

;

commonbase:

         jmp    coldstart

swtuser: jmp    $-$

swtsys:  jmp    $-$

pdisp:   jmp    $-$

xdos:    jmp    $-$

sysdat:  dw     $-$

coldstart:

warmstart:

        mvi     c,0

        jmp     xdos            ; system reset, terminate process

;

;

;I/O handlers

;

;

;  MP/M II V2.0   Console Bios

;

;

nmbcns  equ     3       ; number of consoles

poll    equ     131     ; XDOS poll function

makeque equ     134     ; XDOS make queue function

readque equ     137     ; XDOS read queue function

writeque equ    139     ; XDOS write queue function

xdelay  equ     141     ; XDOS delay function

create  equ     144     ; XDOS create process function

pllpt   equ     0       ; poll printer

plco0   equ     1       ; poll console out #0

plco2   equ     2       ; poll console out #1

plco3   equ     3       ; poll console out #2 (Port 3)

plci3   equ     4       ; poll console in #2 (Port 3)

        if      debug

plci0   equ     5       ; poll console in #0

        endif

;

const:                  ; Console Status

        call    ptbljmp ; compute and jump to hndlr

        dw      pt0st   ; console #0 status routine

        dw      pt2st   ; console #1 (Port 2) status rt

        dw      pt3st   ; console #2 (Port 3) status rt

conin:                  ; Console Input

        call    ptbljmp ; compute and jump to hndlr

        dw      pt0in   ; console #0 input

        dw      pt2in   ; console #1 (Port 2) input

        dw      pt3in   ; console #2 (Port 3) input

conout:                 ; Console Output

        call    ptbljmp ; compute and jump to hndlr

        dw      pt0out  ; console #0 output

        dw      pt2out  ; console #1 (Port 2) output

        dw      pt3out  ; console #2 (Port 3) output

;

ptbljmp:                ; compute and jump to handler

                        ; d = console #

                        ; do not destroy d !

        mov     a,d

        cpi     nmbcns

        jc      tbljmp

        pop     psw     ; throw away table address

rtnempty:

        xra     a

        ret

tbljmp:                 ; compute and jump to handler

                        ; a = table index

        add     a       ; double table index for adr offst

        pop     h       ; return adr points to jump tbl

        mov     e,a

        mvi     d,0

        dad     d       ; add table index * 2 to tbl base

        mov     e,m     ; get handler address

        inx     h

        mov     d,m

        xchg

        pchl            ; jump to computed cns handler

;

; ASCII Character Equates

;

uline   equ     5fh

rubout  equ     7fh

space   equ     20h

backsp  equ     8h

altrub  equ     uline

;

; Input / Output Port Address Equates

;

data0   equ     40h

sts0    equ     data0+1

cd0     equ     sts0

data1   equ     48h

sts1    equ     data1+1

cd1     equ     sts1

data2   equ     50h

sts2    equ     data2+1

cd2     equ     sts2

data3   equ     58h

sts3    equ     data3+1

cd3     equ     sts3

;

; Poll Console #0 Input

;

        if      debug

polci0:

pt0st:

        if      ldcmd

        lda     pt0cntr

        ora     a

        mvi     a,0

        rnz

        endif

        in      sts0

        ani     2

        rz

        mvi     a,0ffh

        ret

;

pt0in:

        if      ldcmd

        lxi     h,pt0cntr

        mov     a,m

        ora     a

        jz      ldcmd0empty

        dcr     m

        lhld    pt0ptr

        mov     a,m

        inx     h

        shld    pt0ptr

        ret

pt0cntr:

        db      ldcmd0empty-pt0ldcmd

pt0ptr:

        dw      pt0ldcmd

pt0ldcmd:

        db      'tod '

ldcmd0empty:

        endif

        mvi     c,poll

        mvi     e,plci0

        call    xdos

        in      data0

        ani     7fh

        ret

;

        else

pt0st:

                        ; return 0ffh if ready,

                        ;        000h if not

        lda     c0inmsgcnt

        ora     a

        rz

        mvi     a,0ffh

        ret

;

; Console #0 Input

;

c0inpd:

        dw      c2inpd  ; pl

        db      0       ; status

        db      32      ; priority

        dw      c0instk+18 ; stkptr

        db      'c0in    '  ; name

        db      0       ; console

        db      0ffh    ; memseg

        ds      36

c0instk:

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      c0inp   ; starting address

c0inq:

        dw      0       ; ql

        db      'c0inque ' ; name

        dw      1       ; msglen

        dw      4       ; nmbmsgs

        ds      8

c0inmsgcnt:

        ds      2       ; msgcnt

        ds      4       ; buffer

c0inqcb:

        dw      c0inq   ; pointer

        dw      ch0in ; msgadr

ch0in:

        db      0

c0inuqcb:

        dw      c0inq   ; pointer

        dw      char0in ; msgadr

char0in:

        db      0

c0inp:

        mvi     c,makeque

        lxi     d,c0inq

        call    xdos    ; make the c0inq

c0inloop:

        mvi     c,flagwait

        mvi     e,6

        call    xdos    ; wait for c0 in intr flag

        mvi     c,writeque

        lxi     d,c0inqcb

        call    xdos    ; write c0in queue

        jmp     c0inloop

pt0in:

                        ; return character in reg A

        mvi     c,readque

        lxi     d,c0inuqcb

        call    xdos            ; read from c0 in queue

        lda     char0in         ; get character

        ani     7fh             ; strip parity bit

        ret

;

        endif

;

; Console #0 Output

;

pt0out:

                        ; Reg C = character to output

        in      sts0

        ani     01h

        jnz     tx0rdy

        push    b

        mvi     c,poll

        mvi     e,plco0

        call    xdos    ; poll console #0 output

        pop     b

tx0rdy:

        mov     a,c

        out     data0

        ret

;

; poll console #0 output

;

polco0:

        in      sts0

        ani     01h

        rz

        mvi     a,0ffh

        ret

;

;

; Line Printer Driver:  TI 810 Serial Printer

;                       TTY Model 40

;

initflag:

        db      0       ; printer initialization flag

list:                   ; List Output

pt1out:

                        ; Reg c = Character to print

        lda     initflag

        ora     a

        jnz     pt1xx

        mvi     a,27h

        out     49h             ; TTY Model 40 init

        sta     initflag

pt1xx:

        in      sts1

        ani     01h

        jnz     tx1rdy

        push    b

        mvi     c,poll

        mvi     e,pllpt

        call    xdos            ; poll printer output

        pop     b

tx1rdy:

        mov     a,c             ; char to register a

        out     data1

        ret

;

; Poll Printer Output

;

pollpt:

                        ; return 0ffh if ready,

                        ;        000h if not

        in      sts1

        ani     01h

        rz

        mvi     a,0ffh

        ret

;

; Poll Console #1 (Port 2) Input

;

pt2st:

                        ; return 0ffh if ready,

                        ;        000h if not

        lda     c2inmsgcnt

        ora     a

        rz

        mvi     a,0ffh

        ret

;

; Console #1 (Port 2) Input

;

c2inpd:

        dw      0       ; pl

        db      0       ; status

        db      34      ; priority

        dw      c2instk+18 ; stkptr

        db      'c2in    '  ; name

        db      2       ; console

        db      0ffh    ; memseg

        ds      36

c2instk:

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h

        dw      c2inp   ; starting address

c2inq:

        dw      0       ; ql

        db      'c2inque ' ; name

        dw      1       ; msglen

        dw      4       ; nmbmsgs

        ds      8

c2inmsgcnt:

        ds      2       ; msgcnt

        ds      4       ; buffer

c2inqcb:

        dw      c2inq   ; pointer

        dw      ch2in   ; msgadr

ch2in:

        db      0

c2inuqcb:

        dw      c2inq   ; pointer

        dw      char2in ; msgadr

char2in:

        db      0

c2inp:

        mvi     c,makeque

        lxi     d,c2inq

        call    xdos    ; make the c2inq

c2inloop:

        mvi     c,flagwait

        mvi     e,8

        call    xdos    ; wait for c2 in intr flag

        mvi     c,writeque

        lxi     d,c2inqcb

        call    xdos    ; write c2in queue

        jmp     c2inloop

pt2in:

                        ; return character in reg A

        mvi     c,readque

        lxi     d,c2inuqcb

        call    xdos            ; read from c2 in queue

        lda     char2in         ; get character

        ani     7fh             ; strip parity bit

        ret

;

; Console #1 (Port 2) Output

;

pt2out:

                        ; Reg C = character to output

        in      sts2

        ani     01h

        jnz     tx2rdy

        push    b

        mvi     c,poll

        mvi     e,plco2

        call    xdos    ; poll console #1 output

        pop     b

tx2rdy:

        mov     a,c

        out     data2

        ret

;

; poll console #1 output

;

polco2:

        in      sts2

        ani     01h

        rz

        mvi     a,0ffh

        ret

;

; Poll Console #2 (Port 3) Input

;

polci3:

pt3st:                  ; return 0ffh if ready,

                        ;        000h if not

        in      sts3

        ani     2

        rz

        mvi     a,0ffh

        ret

;

; Console #2 (Port 3) Input

;

pt3in:                  ; return character in reg A

        mvi     c,poll

        mvi     e,plci3

        call    xdos            ; poll console #0 input

        in      data3           ; read character

        ani     7fh             ; strip parity bit

        ret

;

; Console #2 (Port 3) Output

;

pt3out:                 ; Reg C = character to output

        in      sts3

        ani     01h

        jnz     tx3rdy

        push    b

        mvi     c,poll

        mvi     e,plco3

        call    xdos            ; poll console #2 (Port 3) output

        pop     b

tx3rdy:

        mov     a,c

        out     data3           ; transmit character

        ret

;

; Poll Console #2 (Port 3) Output

;

polco3:

                        ; return 0ffh if ready,

                        ;        000h if not

        in      sts3

        ani     01h

        rz

        mvi     a,0ffh

        ret

;

;

;  MP/M II V2.0   Xios

;

;

polldevice:

                        ; Reg C = device # to be polled

                        ; return 0ffh if ready,

                        ;        000h if not

        mov     a,c

        cpi     nmbdev

        jc      devok

        mvi     a,nmbdev; if dev # >= nmbdev,

                        ; set to nmbdev

devok:

        call    tbljmp  ; jump to dev poll code

devtbl:

        dw      pollpt  ; poll printer output

        dw      polco0  ; poll console #0 output

        dw      polco2  ; poll console #1 output

        dw      polco3  ; poll console #2 output

        dw      polci3  ; poll console #2 input

        if      debug

        dw      polci0  ; poll console #0 input

        endif

nmbdev  equ     ($-devtbl)/2    ; number of devices to poll

        dw      rtnempty; bad device handler

;

; Select / Protect Memory

;

selmemory:

                        ; Reg BC = adr of mem descriptor

                        ; BC ->  base   1 byte,

                        ;        size   1 byte,

                        ;        attrib 1 byte,

                        ;        bank   1 byte.

; this hardware does not have memory protection or

;  bank switching

        ret

;

; Start Clock

;

startclock:

                        ; will cause flag #1 to be set

                        ;  at each system time unit tick

        mvi     a,0ffh

        sta     tickn

        ret

;

; Stop Clock

;

stopclock:

                        ; will stop flag #1 setting at

                        ;  system time unit tick

        xra     a

        sta     tickn

        ret

;

; Exit Region

;

exitregion:

                        ; EI if not preempted or in dispatcher

        lda     preemp

        ora     a

        rnz

        ei

        ret

;

; Maximum Console Number

;

maxconsole:

        mvi     a,nmbcns

        ret

;

; System Initialization

;

systeminit:

;

;  This is the place to insert code to initialize

;  the time of day clock, if it is desired on each

;  booting of the system.

;

        mvi     a,0c3h

        sta     0038h

        lxi     h,inthnd

        shld    0039h           ; JMP INTHND at 0038H

        mvi     c,create

        if      debug

        lxi     d,c2inpd

        else

        lxi     d,c0inpd

        endif

        call    xdos

        lda     intmsk

        out     60h             ; init interrupt mask

        db      0edh,056h       ; Interrupt Mode 1

                                ; ** Z80 Instruction **

        ei

        call    home

        mvi     c,flagwait

        mvi     e,5

        jmp     xdos            ; clear first disk interrupt

;       ret                     ;   & return

;

; Idle procedure

;

;idle:

;       ret

;       -or-

;       ei

;       hlt

;       ret                     ; for full interrupt system

;

;  MP/M II V2.0   Interrupt Handlers

;

flagwait equ    132

flagset equ     133

dsptch  equ     142

inthnd:

                        ; Interrupt handler entry point

                        ;  All interrupts gen a RST 7

                        ;  Location 0038H contains a jmp

                        ;  to INTHND.

        shld    svdhl

        pop     h

        shld    svdret

        push    psw

        lxi     h,0

        dad     sp

        shld    svdsp           ; save users stk ptr

        lxi     sp,lstintstk    ; lcl stk for intr hndl

        push    d

        push    b

        mvi     a,0ffh

        sta     preemp          ; set preempted flag

        in      60h             ; read interrupt mask

        ani     01000000b       ; test & jump if clk int

        jnz     clk60hz

;

        in      stat            ; read disk status port

        ani     08h

        jnz     diskintr

        if      not debug

        in      sts0

        ani     2

        jnz     con0in

        endif

        in      sts2

        ani     2

        jnz     con2in

;       ...                     ; test/handle other ints

;

        jmp     intdone

diskintr:

        xra     a

        out     cmd1            ; reset disk interrupt

        mvi     e,5

        jmp     concmn          ; set flag #5

        if      not debug

con0in:

        in      data0

        sta     ch0in

        mvi     e,6

        jmp     concmn          ; set flag #6

        endif

con2in:

        in      data2

        sta     ch2in

        mvi     e,8

;       jmp     concmn          ; set flag #8

concmn:

        mvi     c,flagset

        call    xdos

        jmp     intdone

clk60hz:

                                ; 60 Hz clock interrupt

        lda     tickn

        ora     a               ; test tickn, indicates

                                ;  delayed process(es)

        jz      notickn

        mvi     c,flagset

        mvi     e,1

        call    xdos            ; set flag #1 each tick

notickn:

        lxi     h,cnt60

        dcr     m               ; dec 60 tick cntr

        jnz     not1sec

        mvi     m,60

        mvi     c,flagset

        mvi     e,2

        call    xdos            ; set flag #2 @ 1 sec

not1sec:

        xra     a

        out     60h

        lda     intmsk

        out     60h             ; ack clock interrupt

;       jmp     intdone

;

;       ...

; Other interrupt handlers

;       ...

;

intdone:

        xra     a

        sta     preemp          ; clear preempted flag

        pop     b

        pop     d

        lhld    svdsp

        sphl                    ; restore stk ptr

        pop     psw

        lhld    svdret

        push    h

        lhld    svdhl

; The following dispatch call will force round robin

;  scheduling of processes executing at the same priority

;  each 1/60th of a second.

; Note: Interrupts are not enabled until the dispatcher

;  resumes the next process.  This prevents interrupt

;  over-run of the stacks when stuck or high frequency

;  interrupts are encountered.

        jmp     pdisp           ; MP/M dispatch

;

;

;       Disk I/O Drivers

;

; Disk Port Equates

;

cmd1    equ     80h

stat    equ     80h

haddr   equ     81h

laddr   equ     82h

cmd2    equ     83h

;

;

home:   ;move to the track o0 position of current drive

        call    headload

; h,l point to word with track for selected disk

homel:

        mvi     m,00    ;set current track ptr back to 0

        in      stat    ;read fdc status

        ani     4       ;test track 0 bit

        rz              ;return if at 0

        stc             ;direction=out

        call    step    ;step one track

        jmp     homel   ;loop

;

seldsk:

        ;drive number in c

        lxi     h,0     ;0000 in hl produces select error

        mov     a,c     ;a is disk number 0 ... ndisks-1

        cpi     ndisks  ;less than ndisks?

        rnc             ;return with HL = 0000 if not

;make sure dummy is 0 (for use in double add to h,l)

        xra     a

        sta     dummy

        mov     a,c

        ani     07h     ;get only disk select bits

        sta     diskno

        mov     c,a

;set up the second command port

        lda     port

        ani     0f0h    ;clear out old disk select bits

        ora     c       ;put in new disk select bits

        ori     08h     ; force double density

        sta     port

;       proper disk number, return dpb element address

        mov     l,c

        dad     h       ;*2

        dad     h       ;*4

        dad     h       ;*8

        dad     h       ;*16

        lxi     d,dpbase

        dad     d       ;HL=.dpb

        shld    tran    ;translate table base

        ret

;

;

;

settrk: ;set track given by register c

        call    headload

;h,l reference correct track indicator according to

;selected disk

        mov     a,c     ;desired track

        cmp     m

        rz              ;we are already on the track

settkx:

        call    step    ;step track-carry has direction

                        ;step will update trk indicator

        mov     a,c

        cmp     m       ;are we where we want to be

        jnz     settkx  ;not yet

;have stepped enough

seekrt:

;need 10 msec delay for final step time and head settle time

        mvi     a,20d

;       call    delay

;       ret             ;end of settrk routine

;

delay:  ;delay for c[A] X .5 milliseconds

        push    b

delay1:

        mvi     c,dlycnst ;constant adjusted to .5 ms loop

delay2:

        dcr     c

        jnz     delay2

        dcr     a

        jnz     delay1

        pop     b

        ret             ;end of delay routine

;

setsec: ;set sector given by register c

        inr     c

        mov     a,c

        sta     sector

        ret

;

sectran:

        ;sector number in c

        ;translate logical to physical sector

        lhld    tran    ;hl=..translate

        mov     e,m     ;E=low(.translate)

        inx     h

        mov     d,m     ;DE=.translate

        mov     a,e     ;zero?

        ora     d       ;00 or 00 = 00

        mvi     h,0

        mov     l,c     ;HL = untranslated sector

        rz              ;skip if so

        xchg

        mov     b,d     ;BC=00ss

        dad     b       ;HL=.translate(sector)

        mov     l,m

        mov     h,d     ;HL=translate(sector)

        ret

;

setdma: ;set dma address given by registers b and c

        mov     l,c     ;low order address

        mov     h,b     ;high order address

        shld    dmaad   ;save the address

        ret

;

;

read:   ;perform read operation.

        ;this is similar to write, so set up read

        ; command and use common code in write

        mvi     b,040h  ;set read flag

        jmp     waitio  ;to perform the actual I/O

;

write:  ;perform a write operation

        mvi     b,080h  ;set write command

;

waitio:

;enter here from read and write to perform the actual

; I/O  operation.  return a 00h in register a if the

; operation completes properly, and 01h if an error

; occurs during the read or write

;

;in this case, the disk number saved in 'diskno'

;                       the track number in 'track'

;                       the sector number in 'sector'

;                       the dma address in 'dmaad'

                        ;b still has r/w flag

        mvi     a,10d   ;set error count

        sta     errors  ;retry some failures 10 times

                        ;before giving up

tryagn:

        push    b

        call    headload

;h,l point to track byte for selected disk

        pop     b

        mov     c,m

; decide whether to allow disk write precompenstation

        mvi     a,39d   ;inhibit precomp on trks 0-39

        cmp     c

        jc      allowit

;inhibit precomp

        mvi     a,10h

        ora     b

        mov     b,a     ;goes out on the same port

                        ; as read/write

allowit:

        lhld    dmaad   ;get buffer address

        push    b       ;b has r/w code   c has track

        dcx     h       ;save and replace 3 bytes below

                        ;buf with trk,sctr,adr mark

        mov     e,m

;figure correct address mark

        lda     port

        ani     08h

        mvi     a,0fbh

        jz      sin

        ani     0fh     ;was double

                        ;0bh is double density

                        ;0fbh is single density

sin:

        mov     m,a

;fill in sector

        dcx     h

        mov     d,m

        lda     sector  ;note that invalid sector number

                        ;will result in head unloaded

                        ;error, so dont check

        mov     m,a

;fill in track

        dcx     h

        pop     b

        mov     a,c

        mov     c,m

        mov     m,a

        mov     a,h     ;set up fdc dma address

        out     haddr   ;high byte

        mov     a,l

        out     laddr   ;low byte

        mov     a,b     ;get r/w flag

        out     cmd1    ;start disk read/write

rwwait:

        push    b

        push    d

        push    h

        mvi     c,flagwait

        mvi     e,5

        call    xdos            ; wait for disk intrpt flag

        pop     h

        pop     d

        pop     b

        mov     m,c     ;restore 3 bytes below buf

        inx     h

        mov     m,d

        inx     h

        mov     m,e

        in      stat    ;test for errors

        ani     0f0h

        rz              ;a will be 0 if no errors

; error from disk

        push    psw     ;save error condition

;check for 10 errors

        lxi     h,errors

        dcr     m

        jnz     redo    ;not ten yet.  do a retry

;we have too many errors. print out hex number for last

;received error type. cpm will print perm error message.

        pop     psw     ;get code

;set error return for operating system

        mvi     a,1

        ret

redo:

;b still has read/write flag

        pop     psw     ;get error code

        ani     0e0h    ;retry if not track error

        jnz     tryagn  ;

;was a track error so need to reseek

        push    b       ;save   read/write indicator

;figure out the desired track

        lxi     d,track

        lhld    diskno  ;selected disk

        dad     d       ;point to correct trk indicator

        mov     a,m     ;desired track

        push    psw     ;save it

        call    home

        pop     psw

        mov     c,a

        call    settrk

        pop     b       ;get read/write indicator

        jmp     tryagn

;

;

;

step:                   ;step head out towards zero

                        ;if carry is set; else

                        ;step in

; h,l point to correct track indicator word

        jc      outx

        inr     m       ;increment current track byte

        mvi     a,04h   ;set direction = in

dostep:

        ori     2

        out     cmd1    ;pulse step bit

        ani     0fdh

        out     cmd1    ;turn off pulse

;the fdc-2 had a stepp ready line. the fdc-3 relies on

;software time out

        mvi     a,16d   ;delay 8 ms

        jmp     delay

;       ret

;

outx:

        dcr     m       ;update track byte

        xra     a

        jmp     dostep

;

headload:

;select and load the head on the correct drive

        lxi     h,prtout        ;old slect info

        mov     b,m

        dcx     h       ;new select info

        mov     a,m

        inx     h

        mov     m,a

        ori     10h     ; enable interrupt

        out     cmd2    ;select the drive

        ani     0efh

;set up h.l to point to track byte for selected disk

        lxi     d,track

        lhld    diskno

        dad     d

;now check for needing a 35 ms delay

;if we have changed drives or if the head is unloaded

;we need to wait 35 ms for head settle

        cmp     b       ;are we on the same drive

        jnz     needdly

;we are on the same drive

;is the head loaded?

        in      stat

        ani     80h

        rz              ;already loaded

needdly:

        xra     a

        out     cmd1    ;load the head

        mvi     a,70d

        jmp     delay

;       ret

;

; BIOS Data Segment

;

cnt60:  db      60      ; 60 tick cntr = 1 sec

intstk:                 ; local intrpt stk

        dw      0c7c7h,0c7c7h,0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h,0c7c7h,0c7c7h

        dw      0c7c7h,0c7c7h,0c7c7h,0c7c7h,0c7c7h

lstintstk:

svdhl:  dw      0       ; saved Regs HL during int hndl

svdsp:  dw      0       ; saved SP during int hndl

svdret: dw      0       ; saved return during int hndl

tickn:  db      0       ; ticking boolean,true = delayed

        if      debug

intmsk: db      44h     ; intrpt msk, enables clk intrpt, & con2

        else

intmsk: db      54h     ; intrpt msk, enables clk intrpt, & con0/2

        endif

preemp: db      0       ; preempted boolean

;

scrat:                  ; start of scratch area

track:  db      0       ; current trk on drive 0

trak1:  db      0       ; current trk on drive 1

trak2:  db      0

trak3:  db      0

sector: db      0       ; currently selected sctr

dmaad:  dw      0       ; current dma address

diskno: db      0       ; current disk number

dummy:  db      0       ; must be 0 for dbl add

errors: db      0

port:   db      0

prtout: db      0

dnsty:  db      0

;

        disks   2

bpb     equ     2*1024  ;bytes per block

rpb     equ     bpb/128 ;records per block

maxb    equ     255     ;max block number

        diskdef 0,1,58,,bpb,maxb+1,128,128,2,0

        diskdef 1,0

;

tran:   ds      2

;

        endef

        db      0       ;force out last byte in hex file

        end

